



Tutorial

EURO Spielplan mit Excel selbst erstellen



Automatische Gruppentabellen

Berücksichtigung aller UEFA-Regeln
einschließlich Fair-Play und Rangliste

Automatische Zuordnung der Teams in der KO-Phase

Wahl der Landessprache

Wahl der Zeitzone

Tabellen der Direkt-Vergleiche

Warnanzeige bei nicht geklärter Reihenfolge in einer Gruppe

https://hermann-baum.de/excel/UEFA_EURO/de/

Inhalt

Zielsetzung	2
1 Struktur des Spielplans	3
2 Auswahl der Sprache	4
3 Auswahl der Zeitzone	8
4 Gruppeneinteilung und Eintragen der Begegnungen (Matches)	10
4.1 Gruppeneinteilung	10
4.2 Begegnungen (Matches)	11
4.3 Anstoßzeiten	12
4.4 Austragungsorte.....	13
5 Gruppenspiele	14
5.1 Formeln für die Spiele.....	14
5.2 Bedingte Formatierungen	15
6 Fair-Play	18
7 Berechnung der Gruppentabellen	20
7.1 Übertragung der Spielergebnisse.....	20
7.2 Rangliste - Teil 1	22
7.3 Das Tabellenblatt Factors.....	25
7.4 Rangliste - Teil 2	26
7.5 Direkter Vergleich dreier Mannschaften.....	28
7.6 Direkter Vergleich zweier Mannschaften.....	34
7.7 Direkter Vergleich aller vier Mannschaften	36
7.8 Rangliste - Teil 3	39
7.9 Gruppentabelle auf dem Spielplan	40
8 KO-Phase	41
8.1 Bestimmung der vier besten Gruppendritten	41
8.2 Achtelfinale	44
8.3 Viertelfinale, Halbfinale und Finale	47
8.4 Hinweis bei Eingabe eines unzulässigen Spielergebnisses	49
9 Optional: Die Sache mit dem roten Punkt	51
9.1 Der rote Punkt.....	51
9.2 Die Meldung.....	52
10 Optional: Zusätzliche Übersicht über die direkten Vergleiche.....	54
10.1 Sichtbarer Bereich - Teil 1	55
10.2 Ausgeblendeter Bereich.....	59
10.3 Sichtbarer Bereich - Teil 2	62
11 Anpassung des Spielplans für eine neue Europameisterschaft	65
11.1 Tabellenblatt 'Language'	65
11.2 Lokalzeit des Gastgeberlandes.....	65
11.3 Zuordnung der Teams zu den Gruppen	65
11.4 Aktualisierung der Match-Daten.....	65
11.5 Aktualisierung der Matchnummern auf dem Spielplan	65
Anhang: Spezialfall ‚Direkte Vergleiche‘	66

Zielsetzung

Der Excel-Spielplan soll folgende Fähigkeiten haben:

1. Automatische Aktualisierung der Gruppentabelle

Nach jedem Eintrag eines Spielergebnisses wird die Gruppentabelle automatisch aktualisiert.

Alle Regeln der UEFA werden berücksichtigt einschließlich Fair-Play und Rangliste.

Die für das Achtelfinale qualifizierten Mannschaften werden farbig hervorgehoben.

2. Automatische Zuordnung der Mannschaften in der KO-Runde

Die Zuordnung der Teams einschließlich der vier besten Gruppendritten erfolgt automatisch.

3. Mehrsprachigkeit

Alle Texte können in den Sprachen englisch, spanisch, italienisch, französisch, deutsch und niederländisch angezeigt werden. Zusätzlich können Übersetzungen in der eigenen Landessprache hinzugefügt werden.

4. Wahl der Zeitzone

Alle Zeitzone der Erde können in einem Auswahlfeld gewählt werden, so dass Datum und Uhrzeit der Spiele in der eigenen lokalen Zeit angezeigt werden.

5. Farbige Hervorhebung der Spiele des aktuellen Tages

Das Datum der Spiele, die am heutigen Tag stattfinden, wird farbig hervorgehoben.

6. Farbige Hervorhebung fehlender Einträge

Haben Spiele in den vergangenen Tagen bereits stattgefunden und sind noch keine Ergebnisse eingetragen, werden die Eingabefelder farbig hervorgehoben.

7. Hinweis, wenn in der KO-Phase ein unzulässiges Unentschieden eingetragen wurde

Die Spiele in der KO-Phase werden in jedem Fall spätestens durch ein Elfmeterschießen entschieden. Wird sowohl für das Spielergebnis als auch für das Elfmeterschießen ein Unentschieden eingetragen, erscheint ein entsprechender Hinweis unter den Eingabefeldern.

8. Anzeige aller direkten Vergleiche in der Gruppenphase

Wenn am Ende der Gruppenphase zwei oder drei Mannschaften punktgleich sind, findet ein direkter Vergleich statt, in dem nur die Spiele dieser Mannschaften gegeneinander berücksichtigt werden. Diese Auswertungen werden auf einem extra Tabellenblatt dargestellt.

9. Einfache Anpassung an zukünftige Europameisterschaften

Die Formeln werden so flexibel gehalten, dass eine Anpassung an das nächste Turnier sehr einfach ist.

Hinweis:

Alle Formeln beziehen sich auf die zehn Exceldateien, die zusätzlich zu diesem Tutorial heruntergeladen werden können (ZIP-Datei). Verglichen mit dem aktuellsten Spielplan kann es geringfügige Abweichungen geben (zum Beispiel durch spätere Änderungen im Tabellenblatt „Language“).

1 Struktur des Spielplans

Im ersten Schritt erstellen wir die Schablone für den Spielplan. Alle Felder bleiben erst einmal leer, weil sie später eine Formel enthalten, die je nach gewählter Sprache den sprachabhängigen Inhalt einfügt. Dieses erste Tabellenblatt nennen wir „EURO“. Es sieht so aus:

The screenshot shows an Excel spreadsheet template for the EURO 2020/2021 tournament. The title "EURO 2020/2021" is centered at the top in orange. The UEFA EURO2020 logo is in the top right corner. The spreadsheet is on a grid with rows 1-87 and columns AA-AD. The layout includes sections for Group A and Group B with match fixtures, final tables (Schlusstabelle), quarterfinals (Viertelfinale), semifinals (Halbfinale), and the final (Finale). A large grey rectangular area is present on the right side of the grid.

Die Eintragungen, die hier schon zu sehen sind, dienen nur zur Illustration. Sie werden später durch Formeln ersetzt.

Die Spalten ab Spalte AA und die Zeilen ab Zeile 94 werden ausgeblendet.

Wer diesen Schritt überspringen möchte, findet das fertige Ergebnis in der Datei „EURO_Tut_1.xlsx“.

2 Auswahl der Sprache

Bevor irgendwo Eintragungen gemacht werden können, brauchen wir ein Tabellenblatt mit den Übersetzungen aller Mannschaftsnamen und aller Beschriftungen in die verschiedenen Landessprachen. Wir erzeugen deshalb ein neues Tabellenblatt mit dem Namen „Language“. Die folgende Abbildung zeigt das Aussehen dieses Tabellenblatts.

Wahl der Sprache										Hier klicken und Sprache wählen:	
	Rank	english	spanish	italian	french	german	dutch	my language		german	
1											
2											
3		english	0								
4		spanish	0								
5		italian	0	1							
6		french	0	2							
7		german	1	3							
8		dutch	0	4							
9		my language	0	5							
10				6							
11				7							
12				8							
13				9							
14				10							
15				11							
16				12							
17				13							
18				14							
19				15							
20				16							
21				17							
22				18							
23				19							
24				20							
25				21							
26				22							
27				23							
28				24							
29				25							
30				26							
31				27							
32				28							
33				29							
34				30							
35				31							
36				32							
37				33							
38				34							
39				35							
40				36							
41				37							
42				38							
43				39							
44				40							
45				41							
46				42							
47				43							
48				44							
49				45							
50				46							
51				47							
52				48							
53				49							
54				50							
55				51							
56				52							
57				53							
58				54							
59				55							
60				0							
61				0							
62				0							
63				0							
64											
Venue no.	Venues										
65											
66	1	London	London	Londres	Londra	Londres	London	Londen			
67	2	Baku	Baku	Bakú	Baku	Bakú	Baku	Baku			
68	3	München	Munich	München	Monaco	Munich	München	München			
69	4	Rom	Rome	Roma	Roma	Rome	Rom	Rom			
70	5	Glasgow	Glasgow	Glasgow	Glasgow	Glasgow	Glasgow	Glasgow			
71	6	Sankt Petersburg	Saint Petersburg	San Petersburgo	San Pietroburgo	Saint-Petersbourg	Sankt Petersburg	St. Petersburg			
72	7	Kopenhagen	Copenhagen	Copenhague	Copenaghen	Copenhague	Kopenhagen	Kopenhagen			
73	8	Sevilla	Seville	Sevilla	Siviglia	Séville	Sevilla	Sevilla			
74	9	Amsterdam	Amsterdam	Amsterdam	Amsterdam	Amsterdam	Amsterdam	Amsterdam			
75	10	Budapest	Budapest	Budapest	Budapest	Budapest	Budapest	Budapest			
76	11	Bukarest	Bucharest	Bucarest	Bucarest	Bucarest	Bukarest	Boekarest			
77	12	Grimbergen	Grimbergen	Grimbergen	Grimbergen	Grimbergen	Grimbergen	Grimbergen			
78	13	Bilbao	Bilbao	Bilbao	Bilbao	Bilbao	Bilbao	Bilbao			
79	14	Dublin	Dublin	Dublin	Dublino	Dublin	Dublin	Dublin			
80	15										
81	16										
82											
Titles											
83		EURO 2020/2021	European Championship 2020/2021	Campeonato de Europa 2020/2021	Campionato Europeo 2020/2021	Championnat européen 2020/2021	EURO 2020/2021	EURO 2020/2021			
84		Direkte Vergleiche	Direct comparisons	Comparaciones directas	Confronti diretti	Comparaisons directes	Direkte Vergleiche	Onderscheidend vergelijk			
85		Fair-Play	Fair play	Fair-play	Fair play	Fair-play	Fair-Play	Eerlijk spel			
86		Wahl der Zeitzone	Choose time zone	Selección de zona horaria	Selezione del fuso orario	Choix du fuseau horaire	Wahl der Zeitzone	Kies tijdzone			
87		Spiele	Matches	Partidos	Partite	Matches	Spiele	Wedstrijden			
88		Wahl der Sprache	Choose language	Elección de la lengua	Scelta della lingua	Choix de langue	Wahl der Sprache	Kies taal			
89		Zuordnung der Gruppendritten im Achtelfinale	Assignment of the group third in the round of 16	Asignación del grupo tercero en los octavos de final	Assegnazione del terzo gruppo agli ottavi di finale	Attribution de la troisième place du groupe en huitième de finale	Zuordnung der Gruppendritten im Achtelfinale	Toewijzing van beste derde plaatsen naar de achtste finales			
90		Gruppen	Groups	Grupos	Gruppi	Groupes	Gruppen	Groepen			
91											

Fortsetzung der Abbildung auf der nächsten Seite

Es enthält die 55 Nationen der UEFA.

Für eine bestimmte Gruppe (z. B. Gruppe A) wählen wir dann später vier Nationen aus, indem wir in die Gruppentabelle vier der laufenden Nummern eintragen, z. B. 14 für die Türkei, 2 für Italien, 19 für Wales und 9 für die Schweiz (siehe Kap. 4.1).

So haben wir maximale Flexibilität auch für zukünftige Spielpläne.

In der Spalte „my language“ kann ein Benutzer alle Übersetzungen in seiner eigenen Muttersprache eintragen und hat dann einen Spielplan komplett in seiner Muttersprache.


Wer sich die Arbeit für das Erstellen dieses Tabellenblatts ersparen möchte, findet das fertige Ergebnis in der Datei „**EURO_Tut_2.xlsx**“.

Im Folgenden wird beschrieben, wie dieses Excel-Blatt erstellt wird und wie es funktioniert.

Zuerst müssen in den Spalten B, D, F, G, H, I, J und K alle Texte eingetragen werden.

Die Spalte D wird eine wichtige Rolle spielen - siehe Kapitel 4.

In der Zelle **N1** wird ein Auswahlfeld eingerichtet: Menü „Daten → Datenüberprüfung“.



Wir wählen die Option „Liste“ aus. Als Quelle für die Listenelemente geben wir den Bereich **\$B\$3:\$B\$9** an. In diesem Bereich tragen wir die Sprachen ein: „english, spanish, italian, ...“.

Nun geht es los mit den ersten Formeln. In die Zelle **C3** tragen wir die Formel **=WENN(\$N\$1=B3;1;0)** ein. Das bedeutet: wenn in der Zelle **N1** die Sprache "english" ausgewählt wurde, erscheint in **C3** eine 1, ansonsten eine 0. Diese Formel kopieren wir nun nach unten bis zur Zelle **C9** (mit der Maus auf die untere rechte Ecke gehen, bis sich der Mauszeiger in ein kleines schwarzes Kreuz verwandelt, dann bei gedrückter linker Maustaste nach unten ziehen).

Jetzt erscheint eine 1 rechts von der Sprache, die im Feld **N1** ausgewählt wurde. Leider kann jemand jedoch auch die Zelle **N1** anklicken und dann die Taste „Entf“ drücken. Dann ist gar keine Sprache ausgewählt. Für diesen Fall soll die Sprache „english“ ausgewählt werden (Default-Wert).

Damit dies der Fall ist, ändern wir die Formel in Zelle **C3**:

=WENN(UND(\$N\$1<>B4;\$N\$1<>B5;\$N\$1<>B6;\$N\$1<>B7;\$N\$1<>B8;\$N\$1<>B9);1;0)

Das bedeutet: Wenn keine der anderen Sprachen ausgewählt ist, dann soll bei „english“ eine 1 stehen.

In die Spalte E tragen wir nun Formeln ein, mit deren Hilfe die Texte der ausgewählten Sprache in die Spalte E kopiert werden.

Wir beginnen mit der Zelle **E5**: **=WENN(\$C\$3;F5;WENN(\$C\$4;G5;WENN(\$C\$5;H5;WENN(\$C\$6;I5;WENN(\$C\$7;J5;WENN(\$C\$8;K5;WENN(L5<>"";L5;F5))))))**

Die Formel beginnt mit **=WENN(\$C\$3;F5;...** Eigentlich müsste da stehen **=WENN(\$C\$3=1;F5;...** Wenn in **C3** eine 1 steht, soll der englische Text aus **F5** in die Spalte C kopiert werden. Da Excel eine 1 als WAHR ansieht und eine 0 als FALSCH, genügt es, für die Bedingung **WENN(\$C\$3;** zu schreiben.

Wenn in **C3** eine 1 steht (**C3** ist WAHR), wird in **E5** der Text von **F5** übernommen, andernfalls gibt es wieder eine WENN-Bedingung: **WENN(\$C\$4;G5;...** Wenn in **C4** eine 1 steht, wird der spanische Text aus Zelle **G5** nach **E5** kopiert usw.

Der letzte Teil der Formel **WENN(L5<>"";L5;F5)** bedeutet: Wenn die Option „my language“ ausgewählt ist und in Spalte L kein Text steht, wird der englische Text verwendet.

Wir kopieren die Formel in **E5** nach unten bis zur Zelle **E160**. In den Zeilen, wo keine Übersetzungen stattfinden, löschen wir die Formeln, also z. B. in **E64, E65, E83**, . Wichtig sind die Dollarzeichen in der Formel: **\$C\$3** soll in der nächsten Zeile unverändert bleiben, aus **F5** soll jedoch **F6, F7**, usw. werden.

Die Hilfsspalten B, C und E können später ausgeblendet werden.

Als ersten Test verbinden wir auf dem Tabellenblatt „EURO“ die Zellen **E1** bis **U1** und tragen die Formel **=Language!\$E\$84** ein. Der Titel unseres Spielplans erscheint nun jeweils in der gewählten Landessprache.

Der aktuelle Stand des Spielplans ist in der Datei „**EURO_Tut_2.xlsx**“ zu finden.

3 Auswahl der Zeitzone

Wir erstellen ein neues Tabellenblatt mit dem Namen „TimeZone“.

	A	B	C	D	E	F	G	H	I	J	K	L	
1	Wahl der Zeitzone						Hier klicken und Zeitzone wählen:	UTC + 2	←				
2													
3													
4		-12	UTC - 12	IDLW		-0,5							
5		-11	UTC - 11			-0,458	Bei Sommerzeit muss die Zeitzone entsprechend angepasst werden (z. B. UTC+2 statt UTC+1)				Zeitzone des Gastgeberlandes:	UTC + 2	
6		-10	UTC - 10	HAST		-0,417							
7		-9	UTC - 9	AKST		-0,375							
8		-8	UTC - 8	PST		-0,333							
9		-7	UTC - 7	MST		-0,292							
10		-6	UTC - 6	CST		-0,25							
11		-5	UTC - 5	EST		-0,208							
12		-4	UTC - 4	AST		-0,167							
13		-3,5	UTC - 3:30	NST		-0,146							
14		-3	UTC - 3			-0,125							
15		-2	UTC - 2			-0,083							
16		-1	UTC - 1			-0,042							
17		0	UTC	WET		0							
18		1	UTC + 1	CET		0,0417							
19		2	UTC + 2	EET/CEST		0,0833							
20		3	UTC + 3	AST/EAT		0,125							
21		3,5	UTC + 3:30	IRT		0,1458							
22		4	UTC + 4			0,1667							
23		4,5	UTC + 4:30			0,1875							
24		5	UTC + 5			0,2083							
25		5,5	UTC + 5:30	IST		0,2292							
26		5,75	UTC + 5:45			0,2396							
27		6	UTC + 6			0,25							
28		6,5	UTC + 6:30			0,2708							
29		7	UTC + 7	ICT		0,2917							
30		8	UTC + 8	CNST		0,3333							
31		9	UTC + 9	JST		0,375							
32		9,5	UTC + 9:30	ACST		0,3958							
33		10	UTC + 10			0,4167							
34		10,5	UTC + 10:30			0,4375							
35		11	UTC + 11			0,4583							
36		12	UTC + 12	IDLE/NZST		0,5							
37		12,75	UTC + 12:45			0,5313							
38		13	UTC + 13			0,5417							
39		14	UTC + 14			0,5833							
40													

Wir tragen in die Zelle **A1** die Formel **=Language!\$E\$87** ein und verbinden die Zellen **A1** bis **F1**.

In die Spalte C tragen wir die zur Zeit existierenden Zeitzonen ein. Wir beginnen mit der Zelle **C4**.

In der Zelle **H1** erzeugen wir ein Auswahlfeld, das genau diese Liste der Zeitzonen enthält (Bereich **\$C\$4:\$C\$39**). Die zusätzlichen Bezeichnungen in Spalte D sind nur Komfort und haben keine weitere Funktionalität.

Entscheidend für die Berechnungen von Datum und Zeit sind die Zahlen im Bereich **\$B\$4:\$B\$39**. Es sind die Verschiebungen in Stunden gegenüber der koordinierten Weltzeit UTC. Die Mitteleuropäische Zeit MEZ oder CET (Central European Time) zum Beispiel ist gleich UTC+1. Diese Zahlen werden nicht durch eine Formel erzeugt, sondern von Hand eingetragen.

Die Anstoßzeiten der einzelnen Spiele werden von der UEFA immer in Ortszeit angegeben. Damit dieser Spielplan auch für zukünftige Europameisterschaften einfach angepasst werden kann, verbinden wir die Zellen **L5** und **L6** und erzeugen hier dasselbe Auswahlfeld wie in **H1**. Hier soll die Zeitzone des Gastgeberlandes eingetragen werden. Die Spalten ab Spalte K werden später ausgeblendet.

Da die EURO 2020/2021 in verschiedenen Ländern ausgetragen wurde, wurden die Anstoßzeiten von der UEFA in MESZ (Mitteleuropäischer Sommerzeit) angegeben. Deshalb wählen wir in diesem zweiten Auswahlfeld die Zeitzone UTC+2 aus.

Wir werden diese Zeitzone in den Umrechnungsformeln für die verschiedenen Ortszeiten berücksichtigen. Auf diese Weise braucht man bei jeder Europameisterschaft nur die offiziellen Anstoßzeiten des Gastgeberlandes einzugeben und hier in dem zweiten Auswahlfeld die Zeitzone des Gastgeberlandes auszuwählen. Alles andere wird dann automatisch berechnet (siehe Kap. 4.2).

In der Spalte B stehen die Stundenverschiebungen in für uns lesbarer Form. Excel speichert bei seinen Datums- und Zeitformaten die Stunden in Bruchteilen eines Tages. Drei Stunden wären also $3/24 = 0,125$.

Ein vollständiges Datum mit Uhrzeit (ein Zeitstempel) ist bei Excel nichts anderes als eine Dezimalzahl, die die Anzahl der Tage angibt, die seit dem 01.01.1900, 00:00 Uhr, vergangen sind. Vor dem Komma stehen die ganzen Tage, die Stellen hinter dem Komma geben den Bruchteil eines Tages an. Die Stellen hinter dem Komma enthalten also die Stunden, Minuten und Sekunden.

Wenn wir zu einem vollständigen Excel-Datum die Dezimalzahl 0,125 addieren, haben wir drei Stunden dazu addiert. Genau so etwas wollen wir ja beim Umrechnen in eine andere Zeitzone erreichen.

In die Zelle **E4** tragen wir die Formel `=B4/24` ein und kopieren sie nach unten bis **E39**. Damit haben wir für jede Zeitzone die entsprechende Dezimalzahl, die wir zu einer UTC-Zeit hinzuaddieren müssen.

Die Zellen **G4:I6** werden miteinander verbunden und wir tragen die Formel `=Language!E144` ein. In Zelle **G1** steht die Formel `=Language!E140` und in den verbundenen Zellen **K5:K6** die Formel `=Language!E141`.

Wir markieren die Zellen **C4** und **D4** und fügen eine bedingte Formatierung mit der Formel `=H$1=$C4` ein. Die ausgewählte Zeitzone soll grün eingefärbt sein. Die Formatierung wird mit „Format übertragen“ auf alle Zeitzonen in den Spalten C und D übertragen.

Die Spalten B und E werden später ausgeblendet.

Der aktuelle Stand des Spielplans ist in der Datei „**EURO_Tut_3.xlsx**“ zu finden.

4 Gruppeneinteilung und Eintragen der Begegnungen (Matches)

Der einfachste Weg wäre, die Begegnungen in den einzelnen Gruppen direkt auf dem Tabellenblatt „EURO“ einzutragen. Um ein Maximum an Flexibilität zu erreichen und eine Anpassung des Spielplans für zukünftige Turniere möglichst bequem zu machen, werden wir zwei Zwischenschritte einbauen.

Im ersten Schritt nehmen wir auf einem extra Tabellenblatt mit dem Namen „**Groups**“ die Gruppeneinteilung vor. Im zweiten Schritt erstellen wir auf einem Tabellenblatt mit dem Namen „**Matches**“ eine Übersicht mit den 51 Begegnungen des Turniers. Die Übertragung auf das Tabellenblatt „EURO“ erfolgt dann automatisch mit Hilfe von Formeln.

4.1 Gruppeneinteilung

Wir erstellen ein neues Tabellenblatt mit dem Namen „**Groups**“.

	A	B	C	D	E
1					
2			Gruppen		
3					
4		Team	Team no.	Name	
5				A	
6					
7		A1	14	Türkei	
8		A2	2	Italien	
9		A3	19	Wales	
10		A4	9	Schweiz	
11				B	
12		B1	15	Dänemark	
13		B2	20	Finnland	
14		B3	1	Belgien	
15		B4	12	Russland	
16				C	
17		C1	11	Niederlande	
18		C2	6	Ukraine	
19		C3	16	Österreich	
20		C4	30	Nordmazedonien	
21				D	
22		D1	3	England	
23		D2	10	Kroatien	
24		D3	29	Schottland	
25		D4	18	Tschechien	
26				E	
27		E1	5	Spanien	
28		E2	17	Schweden	
29		E3	8	Polen	
30		E4	22	Slowakei	
31				F	
32		F1	31	Ungarn	
33		F2	13	Portugal	
34		F3	7	Frankreich	
35		F4	4	Deutschland	
36					

Auf dem Tabellenblatt „Language“ sind bereits die Namen der Nationen in verschiedenen Sprachen vorhanden. Auf dem Tabellenblatt „Groups“ soll nun die Gruppeneinteilung stattfinden.

In der Spalte B tragen wir die Abkürzungen A1, A2, ... F4 ein. In die Spalte C tragen wir die Nummern aus der Nationen-Liste des Tabellenblatts „Language“ ein, also 14 für die Türkei, 2 für Italien, 19 für Wales, 9 für die Schweiz usw.

Bei einer neuen Gruppeneinteilung müssen einfach nur diese Nummern geändert werden.

Damit in Spalte D automatisch die zugehörigen Namen erscheinen, tragen wir in Zelle **D7** die folgende Formel ein:

=SVERWEIS(\$C7;Language!\$D\$5:\$E\$63;2;0)

Die Funktion **SVERWEIS** hat vier Parameter.

1. Parameter: **\$C7**

Nach dem Inhalt von **C7** wird gesucht, in unserem Beispiel also nach „14“.

2. Parameter: **Language!\$D\$5:\$E\$63**

Das ist der Bereich, in dem gesucht und gefunden wird. Die Suchspalte ist immer die erste Spalte ganz links.

3. Parameter: **2**

Das Ergebnis soll in der zweiten Spalte gefunden werden.

4. Parameter: **0**

Die „0“ (oder auch „FALSCH“) bewirkt, dass beim Suchen in der Suchspalte nur bei genauer Übereinstimmung gefunden wird.

► [Ausführliche Erklärung der Funktion SVERWEIS](#)

Die Formel schaut also in unserer Nationen-Liste nach, welcher Name zur Nummer 14 gehört und trägt diesen Namen in die Zelle **D7** ein.

Wir kopieren die Formel nach unten bis zur Zelle **D35**.

4.2 Begegnungen (Matches)

Wir erstellen ein neues Tabellenblatt mit dem Namen „Matches“.

Spiele											
Match No.	Teams		Date (local time host)		Date (local time)		Venue no.	Venue	Team 1	Team 2	Order of headings
1	A1	A2	Fr, 11.06.2021	21:00	Fr, 11.06.2021	21:00	1	London	Türkei	Italien	
2	A3	A4	Sa, 12.06.2021	15:00	Sa, 12.06.2021	15:00	2	Baku	Wales	Schweiz	
3	B1	B2	Sa, 12.06.2021	18:00	Sa, 12.06.2021	18:00	7	Kopenhagen	Dänemark	Finnland	
4	B3	B4	Sa, 12.06.2021	21:00	Sa, 12.06.2021	21:00	6	Sankt Ptersburg	Belgien	Russland	
5	C1	C2	So, 13.06.2021	18:00	So, 13.06.2021	18:00	11	Bukarest	Niederlande	Ukraine	
6	C3	C4	So, 13.06.2021	21:00	So, 13.06.2021	21:00	9	Amsterdam	Österreich	Nordmazedonien	
7	D1	D2	So, 13.06.2021	15:00	So, 13.06.2021	15:00	1	London	England	Kroatien	
8	D3	D4	Mo, 14.06.2021	15:00	Mo, 14.06.2021	15:00	5	Glasgow	Schottland	Tschechien	
9	E1	E2	Mo, 14.06.2021	18:00	Mo, 14.06.2021	18:00	6	Sankt Ptersburg	Spanien	Schweden	
10	E3	E4	Mo, 14.06.2021	21:00	Mo, 14.06.2021	21:00	8	Sevilla	Polen	Slowakei	
11	F1	F2	Di, 15.06.2021	18:00	Di, 15.06.2021	18:00	10	Budapest	Ungarn	Portugal	
12	F3	F4	Di, 15.06.2021	21:00	Di, 15.06.2021	21:00	3	München	Frankreich	Deutschland	
13	A1	A3	Mi, 16.06.2021	18:00	Mi, 16.06.2021	18:00	2	Baku	Türkei	Wales	
14	A2	A4	Mi, 16.06.2021	21:00	Mi, 16.06.2021	21:00	1	London	Italien	Schweiz	
15	B1	B3	Do, 17.06.2021	18:00	Do, 17.06.2021	18:00	7	Kopenhagen	Dänemark	Belgien	
16	B2	B4	Mi, 16.06.2021	15:00	Mi, 16.06.2021	15:00	6	Sankt Ptersburg	Finnland	Russland	
17	C1	C3	Do, 17.06.2021	21:00	Do, 17.06.2021	21:00	9	Amsterdam	Niederlande	Österreich	
18	C2	C4	Do, 17.06.2021	15:00	Do, 17.06.2021	15:00	11	Bukarest	Ukraine	Nordmazedonien	
19	D1	D3	Fr, 18.06.2021	21:00	Fr, 18.06.2021	21:00	1	London	England	Schottland	
20	D2	D4	Fr, 18.06.2021	18:00	Fr, 18.06.2021	18:00	5	Glasgow	Kroatien	Tschechien	
21	E1	E3	Sa, 19.06.2021	21:00	Sa, 19.06.2021	21:00	8	Sevilla	Spanien	Polen	
22	E2	E4	Fr, 18.06.2021	15:00	Fr, 18.06.2021	15:00	6	Sankt Ptersburg	Schweden	Slowakei	
23	F1	F3	Sa, 19.06.2021	15:00	Sa, 19.06.2021	15:00	10	Budapest	Ungarn	Frankreich	
24	F2	F4	Sa, 19.06.2021	18:00	Sa, 19.06.2021	18:00	3	München	Portugal	Deutschland	
25	A4	A1	So, 20.06.2021	18:00	So, 20.06.2021	18:00	2	Baku	Schweiz	Türkei	
26	A2	A3	So, 20.06.2021	18:00	So, 20.06.2021	18:00	1	London	Italien	Wales	
27	B4	B1	Mo, 21.06.2021	21:00	Mo, 21.06.2021	21:00	7	Kopenhagen	Russland	Dänemark	
28	B2	B3	Mo, 21.06.2021	21:00	Mo, 21.06.2021	21:00	6	Sankt Ptersburg	Finnland	Belgien	
29	C4	C1	Mo, 21.06.2021	18:00	Mo, 21.06.2021	18:00	11	Bukarest	Nordmazedonien	Niederlande	
30	C2	C3	Mo, 21.06.2021	18:00	Mo, 21.06.2021	18:00	9	Amsterdam	Ukraine	Österreich	
31	D4	D1	Di, 22.06.2021	21:00	Di, 22.06.2021	21:00	1	London	Tschechien	England	
32	D2	D3	Di, 22.06.2021	21:00	Di, 22.06.2021	21:00	5	Glasgow	Kroatien	Schottland	
33	E4	E1	Mi, 23.06.2021	18:00	Mi, 23.06.2021	18:00	6	Sankt Ptersburg	Slowakei	Spanien	
34	E2	E3	Mi, 23.06.2021	18:00	Mi, 23.06.2021	18:00	8	Sevilla	Schweden	Polen	
35	F4	F1	Mi, 23.06.2021	21:00	Mi, 23.06.2021	21:00	3	München	Deutschland	Ungarn	
36	F2	F3	Mi, 23.06.2021	21:00	Mi, 23.06.2021	21:00	10	Budapest	Portugal	Frankreich	
1/8 finals											
37	2A	2B	Sa, 26.06.2021	18:00	Sa, 26.06.2021	18:00	9	Amsterdam			
38	1A	2C	Sa, 26.06.2021	21:00	Sa, 26.06.2021	21:00	1	London			
39	1C	3D/E/F	So, 27.06.2021	18:00	So, 27.06.2021	18:00	10	Budapest			
40	1B	3A/D/E/F	So, 27.06.2021	21:00	So, 27.06.2021	21:00	8	Sevilla			
41	2D	2E	Mo, 28.06.2021	18:00	Mo, 28.06.2021	18:00	7	Kopenhagen			
42	1F	3A/B/C	Mo, 28.06.2021	21:00	Mo, 28.06.2021	21:00	11	Bukarest			
43	1D	2F	Di, 29.06.2021	18:00	Di, 29.06.2021	18:00	1	London			
44	1E	3A/B/C/D	Di, 29.06.2021	21:00	Di, 29.06.2021	21:00	5	Glasgow			
1/4 finals											
45	W41	W42	Fr, 02.07.2021	18:00	Fr, 02.07.2021	18:00	6	Sankt Ptersburg			
46	W40	W38	Fr, 02.07.2021	21:00	Fr, 02.07.2021	21:00	3	München			
47	W39	W37	Sa, 03.07.2021	18:00	Sa, 03.07.2021	18:00	2	Baku			
48	W43	W44	Sa, 03.07.2021	21:00	Sa, 03.07.2021	21:00	4	Rom			
1/2 finals											
49	W45	W46	Di, 06.07.2021	21:00	Di, 06.07.2021	21:00	1	London			
50	W47	W48	Mi, 07.07.2021	21:00	Mi, 07.07.2021	21:00	1	London			
Final											
51	W49	W50	So, 11.07.2021	21:00	So, 11.07.2021	21:00	1	London			

Die Idee ist folgende:

Wir erstellen hier auf dem Tabellenblatt „Matches“ eine Übersicht über alle 51 Spiele. Dabei nummerieren wir die Spiele von 1 bis 51 durch. Diese Nummern werden in Spalte B eingetragen. Die Reihenfolge wird dabei von dem üblichen Schema der UEFA für den Ablauf des Turniers bestimmt: zuerst die Gruppenspiele A1-A2, A3-A4, B1-B2, B3-B4 usw. bis H2-H3, danach die KO-Spiele.

21	Sevilla		
	Sa, 19.06.2021	21:00	
	Spanien	Polen	
	1	1	

Auf dem Tabellenblatt „EURO“, unserem eigentlichen Spielplan, gibt es für jedes Spiel ein Kästchen mit Spielort, Datum, Teams und Ergebnis. Jedes dieser Kästchen bekommt links oben eine dieser Nummern, z. B. die Nr. 21. Die Formeln in dem Kästchen sorgen dann dafür, dass die Daten des Spiels Nr. 21 in dem Kästchen angezeigt werden.

Bei der Anpassung an eine neue EM können wir nun frei entscheiden, wo wir das Spiel Nr. 21 auf dem neuen Spielplan platzieren.

In die Spalten C und D des Tabellenblattes „Matches“ übertragen wir zunächst die Teams nach dem von der UEFA veröffentlichten Spielplan. Spiel Nr. 21 wäre also z. B. E1-E3, das heißt Mannschaft 1 der Gruppe E gegen Mannschaft 3 der Gruppe E.

In die Spalten I und J tragen wir nun nicht die Namen direkt ein, sondern eine Formel, die uns die Namen automatisch liefert. In der Zelle **C4** steht **A1**. Wir müssen also für das Match Nr. 1 den Namen des ersten Teams auf dem Blatt "Groups" unter **A1** nachschauen.

Diesen Namen finden wir in der Zelle **Groups!D7**.

Für ein solches Nachschlagen gibt es in Excel die Funktion **SVERWEIS**. Deshalb tragen wir auf dem Tabellenblatt "Matches" folgende Formel in die Zelle **I4** ein: **=SVERWEIS(C4;Groups!\$B\$7:\$D\$35;3;0)**

Wichtig ist, dass der erste Parameter keine Dollarzeichen enthält. Somit können wir diese Formel nach rechts und nach unten gleichzeitig kopieren. Wir gehen auf die rechte untere Ecke der Zelle **I4**, bis sich der Cursor in ein kleines schwarzes Kreuz verwandelt, und ziehen die Formel über den gesamten Bereich **I4:J39**. Jetzt erscheinen in diesem Bereich die Mannschaftsnamen in der aktuell gewählten Sprache.

4.3 Anstoßzeiten

In die Spalte E tragen wir Datum und Uhrzeit für jedes Spiel ein und formatieren die Zellen so, dass Datum und Uhrzeit angezeigt werden.

In der Spalte F sollen nun die Zeiten in die Zeitzone umgerechnet werden, die im Feld **TimeZone!H1** ausgewählt wurde.

Wir tragen in **F4** die folgende Formel ein:

```
=E4-SVERWEIS(TimeZone!$L$5;TimeZone!$C$4:$E$39;3;0)
+SVERWEIS(TimeZone!$H$1;TimeZone!$C$4:$E$39;3;0)
```

Die erste SVERWEIS-Funktion schlägt in der Tabelle aller Zeitzonen nach, welche Dezimalzahl (positiv oder negativ) zur Ortszeit des Gastgeberlandes gehört. Das ist im Beispiel Deutschland (EURO 2024) die Zahl $2/24 \approx 0,0833$.

Von der Zeit am Austragungsort (**E4**) wird zuerst diese Zahl (0,0833) subtrahiert. Es werden zwei Stunden subtrahiert. Damit wird also auf UTC umgerechnet. Mit dem Ausdruck

SVERWEIS(TimeZone!\$H\$1;TimeZone!\$C\$4:\$E\$39;3;0) wird nun in der Tabelle aller Zeitzonen nachgeschlagen, welche Dezimalzahl (positiv oder negativ) noch addiert werden muss, um die gewünschte lokale Zeit zu erhalten.

Wir kopieren die Formel in **F4** nach unten bis **F58**.

In den Zellen **F40**, **F49**, **F54** und **F57** löschen wir die Formel wieder.

4.4 Austragungsorte

Am bequemsten ist es, wenn wir für jedes Spiel nur die Nummer des Austragungsortes eintragen müssen. Diese Nummern schauen wir im Tabellenblatt „**Language**“ in der Spalte **D** nach und tragen sie auf dem Tabellenblatt "**Matches**" in die Spalte **G** ein.

In die Zelle **H4** geben wir nun die Formel **=WENNNV(SVERWEIS(\$G4;Language!\$D\$97:\$E\$112;2;0);"")** ein und kopieren sie nach unten bis zur Zelle **H58**. In den Zeilen 40, 49, 54 und 57 löschen wir die Formel wieder. Damit erscheinen die Namen der Austragungsorte in der Spalte H.

Die Funktion **WENNNV** hat zwei Parameter:

- 1. Parameter:** der Wert, der in der Zelle erscheinen soll
- 2. Parameter:** das, was im Falle des Fehlers "#NV" (no value) in der Zelle erscheinen soll.

In unserem Fall bleibt die Zelle leer, wenn die SVERWEIS-Funktion nichts findet.

Die Spalten B, F, H, I und J sollten mit Hilfe eines Blattschutzes gesperrt werden, um ein versehentliches Überschreiben zu vermeiden. Sie werden nicht mehr verändert.

Die Spalten C, D, E und G dagegen werden bei der nächsten Europameisterschaft neu angepasst.

Der aktuelle Stand des Spielplans ist in der Datei „**EURO_Tut_4.xlsx**“ zu finden.

5 Gruppenspiele

Wir werden für das erste Gruppenspiel die erforderlichen Formeln eintragen und diese Formeln dann durch Kopieren auf alle anderen Spiele der Gruppenphase und auch des Achtelfinales übertragen. Die Formeln von Viertelfinale, Halbfinale und Finale werden später besprochen und eingetragen (siehe Kap. 8).

Mit Hilfe der bedingten Formatierung werden wir die Datumsfelder, in denen das heutige Datum steht, farbig hervorheben. So sieht man sofort, welche Spiele am heutigen Tag stattfinden.

Ebenfalls mit Hilfe der bedingten Formatierung werden wir leere Eingabefelder von Spielen, die in der Vergangenheit liegen, farbig hervorheben. So sieht man sofort, welche Spielergebnisse noch nachzutragen sind.

5.1 Formeln für die Spiele

Wir gehen zum Tabellenblatt „EURO“ und tragen die folgenden Formeln ein:

Zelle B9: `=Language!E93&" A"`

Zelle E9: `=Language!E93&" B"`

usw. bis zur Gruppe F. Somit erscheinen diese Überschriften immer in der gewählten Landessprache.

Wir haben alle Spiele von 1 bis 51 durchnummeriert. Diese Nummern tragen wir nun nach dem folgenden Schema links oben neben dem Kästchen mit den Spieldaten ein:

Gruppe A	Gruppe B	Gruppe C	Gruppe D	Gruppe E	Gruppe F
1	3	5	7	9	11
2	4	6	8	10	12
13	15	17	19	21	23
14	16	18	20	22	24
25	27	29	31	33	35
26	28	30	32	34	36

Das erste Spiel in Gruppe A ist das Spiel Nr. 1. Wir tragen also in der Zelle A11 eine 1 ein. In die Zelle A16 tragen wir eine 2 ein usw.

Achtelfinale: 41, 42, 40, 38, 39, 37, 43, 44 (Erläuterung der Reihenfolge in Kap. 8)

Viertelfinale: 45, 46, 47, 48

Halbfinale: 49, 50

Finale: 51

In Zelle A52 kommt also eine 41, in Zelle D52 eine 42, in Zelle G52 eine 40 usw.

Nun werden wir die Kästchen für die Spiele der Gruppenphase und des Achtelfinales mit Inhalt füllen. Normalerweise würden wir auf dem Tabellenblatt „Matches“ nachschauen, welche Daten zum Spiel Nr. 1 gehören und diese Daten in das Kästchen übertragen. Mit Hilfe der Funktion SVERWEIS lassen wir diese Übertragung der Daten automatisch ablaufen.

In die Zelle **B11** tragen wir die Formel **=SVERWEIS(A11;Matches!\$B\$4:\$K\$58;7;0)** ein. Das Suchkriterium in dieser Formel ist **A11**. In **A11** steht eine 1.

Diese 1 wird auf dem Tabellenblatt „Matches“ in der ersten Spalte des Bereichs **B\$4:\$K\$58** gesucht. Nachdem sie in der Zeile 4 des Tabellenblattes gefunden wurde, wird der Inhalt in der Spalte 7 („London“) von der Funktion SVERWEIS zurückgegeben. Als Austragungsort erscheint in der Zelle **B11** nun wie gewünscht „London“.

Alle weiteren Formeln haben denselben Aufbau:

Zelle **B12**: **=SVERWEIS(A11;Matches!\$B\$4:\$K\$58;5;0)**

Zelle **B13**: **=SVERWEIS(A11;Matches!\$B\$4:\$K\$58;8;0)**

Zelle **C13**: **=SVERWEIS(A11;Matches!\$B\$4:\$K\$58;9;0)**


A11 ist ein relativer Bezug (ohne Dollarzeichen). Deshalb können wir auf dem Tabellenblatt „EURO“ den Bereich **B11:C13** markieren und in die Kästchen für die anderen Gruppenspiele kopieren. In jedem Kästchen erscheinen nun die Daten des Spiels mit der entsprechenden Nummer. Wir probieren es aus: wir ändern z. B. in Zelle **M16** die Nummer von 10 in 9. Sofort sehen wir das Spiel Nr. 9 in diesem Kästchen.

Diese Flexibilität erleichtert uns auch die Anordnung der Spiele für das Achtelfinale (siehe Kap. 8).


5.2 Bedingte Formatierungen

Wir klicken auf die verbundenen Zellen **B12:C12** und wählen im Menü „Start → Bedingte Formatierung → Neue Regel“. Als Regeltyp wählen wir die letzte Option und tragen die Formel **=GANZZAHL(B12)=HEUTE()** ein:


Wir klicken auf „Formatieren...“ und wählen einen gelben Hintergrund. Wir klicken auf OK.



Wir sehen nun das folgende Fenster:



Ein Klick auf den schwarzen Pfeil-Nach-Oben öffnet ein Adressen-Fenster mit der Liste aller Adressen, auf welche die Regel angewendet wird.



Im Spielplan klicken wir nun bei gedrückter STRG-Taste auf alle weiteren Datumfelder der Gruppenphase und der KO-Runde. Anschließend klicken wir auf den Pfeil-Nach-Unten im Adressen-Fenster und schließen den Vorgang mit einem Klick auf OK ab.

Die bedingte Formatierung der Eingabefelder für die Spielergebnisse erfordert drei Formeln:

Zelle **B14**: **=UND(B12<HEUTE();B14="")**

Die bedingte Formatierung wird auf alle **links** stehenden Eingabefelder der Gruppenphase und der KO-Runde erweitert, also E14, H14 usw.

Zelle **C14**: **=UND(B12<HEUTE();C14="")**


Die bedingte Formatierung wird auf alle **rechts** stehenden Eingabefelder der Gruppenphase und des Achtelfinales erweitert, also F14, I14 usw.

Zelle **E65**: **=UND(C63<HEUTE();E65="")**

Erweiterung auf alle **rechts** stehenden Eingabefelder der restlichen KO-Spiele (Viertelfinale, Halbfinale und Finale).

Mit Hilfe der **Datenüberprüfung** verhindern wir unsinnige Eingaben für die Spielergebnisse.

Wir markieren alle Eingabefelder und wählen im Menü Daten → Datenüberprüfung.



Wir lassen nur ganze Zahlen zwischen 0 und 99 zu.

Der aktuelle Stand des Spielplans ist in der Datei „**EURO_Tut_5.xlsx**“ zu finden.

6 Fair-Play

Sollte der Fall eintreten, dass zwei Mannschaften am Ende der Gruppenphase Gleichstand in der Punktzahl, der Tordifferenz und der Anzahl der erzielten Tore haben und auch ihre Direktbegegnung unentschieden endete, entscheidet die Fair-Play-Wertung oder das Ranking in der Qualifikationsrunde über die bessere Platzierung.

Für diesen Fall genügt es, wenn man bei der begünstigten Mannschaft einen Bonus-Punkt einträgt, der diese Mannschaft dann an die bessere Position bringt. Alternativ kann man auch negative Werte eintragen.

Auf einem neuen Tabellenblatt mit dem Namen „FairPlay“ ermöglichen wir diese Eingabe eines Bonus-Punktes.

	A	B	C	D	E	F	G
1							
2	Fair-Play						
3	Entscheidet bei zwei Mannschaften die Fair-Play-Wertung über die bessere Platzierung in der Gruppe, so wird hier für die bessere Mannschaft eine "1" eingetragen.						
4							
5	Gruppe A	Bonus			Gruppe D	Bonus	
6	Türkei	0			England	0	
7	Italien	0			Kroatien	0	
8	Wales	0			Schottland	0	
9	Schweiz	0			Tschechien	0	
10							
11	Gruppe B	Bonus			Gruppe E	Bonus	
12	Dänemark	0			Spanien	0	
13	Finnland	0			Schweden	0	
14	Belgien	0			Polen	0	
15	Russland	0			Slowakei	0	
16							
17	Gruppe C	Bonus			Gruppe F	Bonus	
18	Niederlande	0			Ungarn	0	
19	Ukraine	0			Portugal	0	
20	Österreich	0			Frankreich	0	
21	Nordmazedonien	0			Deutschland	0	
22							

Zelle B2: `=Language!E86`

Zelle B3: `=Language!E139`

In die Zellen links neben den Überschriften - also in A5, A11, A17, D5, D11 und D17 - tragen wir den Buchstaben der betreffenden Gruppe ein und blenden ihn mit Hilfe der benutzerdefinierten Formatierung „;;;“ (drei Semikolon) aus.

Zelle B5: `=Language!E93&" "&A5`

Zelle C5: `=Language!E124`

Zelle **B6**: **=SVERWEIS(A5&"1";Groups!\$B\$7:\$D\$35;3;0)**

Zelle **B7**: **=SVERWEIS(A5&"2";Groups!\$B\$7:\$D\$35;3;0)**

Zelle **B8**: **=SVERWEIS(A5&"3";Groups!\$B\$7:\$D\$35;3;0)**

Zelle **B9**: **=SVERWEIS(A5&"4";Groups!\$B\$7:\$D\$35;3;0)**

Die letzten vier Formeln lesen aus dem Tabellenblatt „Groups“ die vier Mannschaftsnamen.

In die Zellen **C6** bis **C9** tragen wir Nullen ein.

Nun markieren wir den Bereich **B5:C9** und kopieren ihn fünf Mal an die Positionen B11, B17, E5, E11 und E17.

Der aktuelle Stand des Spielplans ist in der Datei „**EURO_Tut_6.xlsx**“ zu finden.

7 Berechnung der Gruppentabellen

Das Ziel ist es, auf unserem Spielplan in dem Bereich **B3:C6** den aktuellen Stand der Gruppentabelle anzuzeigen und die Reihenfolge der Mannschaften automatisch in die Schlusstabelle **B46:B49** einzutragen.

Wir erzeugen ein neues Tabellenblatt und nennen es „GrpA“. Es wird so aussehen:

Die anderen Tabellenblätter mit den Namen „GrpB“, „GrpC“, usw. erzeugen wir dann so, dass wir **sieben Kopien des Tabellenblattes „GrpA“ anfertigen** und lediglich in der Zelle **C1** ein „B“ bzw. ein „C“ usw. eintragen!

7.1 Übertragung der Spielergebnisse

Die Spielergebnisse für Gruppe A finden wir auf dem Spielplan in den Spalten B und C, diejenigen für Gruppe B in den Spalten E und F usw. Auf den Tabellenblättern „GrpA“, „GrpB“, usw. sollen in den Zellen mit gelbem Hintergrund automatisch die Mannschaften und Spielergebnisse der richtigen Gruppe erscheinen, je nachdem, welcher Buchstabe in der Zelle **C1** eingetragen ist.

Die Formel in Zelle **B4** könnte einfach lauten: **=EURO!B13**. Dann wäre diese Formel aber nur für die Gruppe A richtig. Deshalb werden wir diese Formel mit Hilfe der Funktion INDEX flexibel gestalten.

Wir tragen in Zelle **B4** die folgende Formel ein: **=INDEX(EURO!\$B\$13:\$R\$40;1;1+(CODE(\$C\$1)-65)*3)**.

Die Funktion INDEX hat folgende Syntax:

INDEX(Bereich; Zeile; [Spalte]).

Die Parameter ‚Zeile‘ und ‚Spalte‘ geben an, an welcher Stelle des angegebenen Bereiches ein Wert herausgepickt wird.

Die Zeilenindices für die sechs Gruppenspiele sind 1, 6, 11, 16, 21 und 28. Der Parameter ‚Spalte‘ wird so berechnet, dass jeweils auf die richtige Gruppe zugegriffen wird.

Die Funktion **CODE(\$C\$1)** liefert den ASCII-Code des Buchstabens aus Zelle **C1**, z.B. **67** für ein ‚C‘. Der Rechenterm **(1+CODE(\$C\$1)-65)*3** berechnet den richtigen Spaltenindex für die jeweilige Gruppe, z.B. **7** für die Gruppe C.

Wir kopieren die Formel in Zelle **B4** nach unten, korrigieren aber in jeder Zeile den Parameter ‚**Zeile**‘ manuell:

Zelle **B4**: =INDEX(EURO!\$B\$13:\$R\$40;1;1+(CODE(\$C\$1)-65)*3)

Zelle **B5**: =INDEX(EURO!\$B\$13:\$R\$40;6;1+(CODE(\$C\$1)-65)*3)

Zelle **B6**: =INDEX(EURO!\$B\$13:\$R\$40;11;1+(CODE(\$C\$1)-65)*3)

Zelle **B7**: =INDEX(EURO!\$B\$13:\$R\$40;16;1+(CODE(\$C\$1)-65)*3)

Zelle **B8**: =INDEX(EURO!\$B\$13:\$R\$40;21;1+(CODE(\$C\$1)-65)*3)

Zelle **B9**: =INDEX(EURO!\$B\$13:\$R\$40;28;1+(CODE(\$C\$1)-65)*3)

Die sechs Formeln im Bereich **D4:D9** sind fast identisch. Nur der Parameter ‚**Spalte**‘ beginnt jeweils mit ‚**2+(CODE...**‘.

Damit die Anzahl der Tore nur bei den Spielen berücksichtigt wird, bei denen beide Felder für die Tore ausgefüllt sind, haben die Formeln in den Spalten E und G die Form:

Zelle **E4**: =WENN(ODER(Tore1="";Tore2="");"";Tore1)

Zelle **G4**: =WENN(ODER(Tore1="";Tore2="");"";Tore2)

Die Formeln lauten:

Zelle **E4**: =WENN(ODER(INDEX(EURO!\$B\$14:\$R\$41;1;1+(CODE(\$C\$1)-65)*3)="";
INDEX(EURO!\$B\$14:\$R\$41;1;2+(CODE(\$C\$1)-65)*3)=""");"";
INDEX(EURO!\$B\$14:\$R\$41;1;1+(CODE(\$C\$1)-65)*3))

Zelle **G4**: =WENN(ODER(INDEX(EURO!\$B\$14:\$R\$41;1;1+(CODE(\$C\$1)-65)*3)="";
INDEX(EURO!\$B\$14:\$R\$41;1;2+(CODE(\$C\$1)-65)*3)=""");"";
INDEX(EURO!\$B\$14:\$R\$41;1;1+(CODE(\$C\$1)-65)*3))

Die roten Zeilenindices werden auch hier in den Zellen **E5:E9** und **G5:G9** jeweils entsprechend angepasst.

Auch die Formeln für die Erfassung des Elfmeterschießens haben dieselbe Struktur:

Zelle **J8**: =WENN(ODER(INDEX(EURO!\$B\$36:\$R\$36;1;1+(CODE(\$C\$1)-65)*3)="";
INDEX(EURO!\$B\$36:\$R\$36;1;2+(CODE(\$C\$1)-65)*3)=""");"";
INDEX(EURO!\$B\$36:\$R\$36;1;1+(CODE(\$C\$1)-65)*3))

Zelle **J9**: =WENN(ODER(INDEX(EURO!\$B\$43:\$R\$43;1;1+(CODE(\$C\$1)-65)*3)="";
INDEX(EURO!\$B\$43:\$R\$43;1;2+(CODE(\$C\$1)-65)*3)=""");"";
INDEX(EURO!\$B\$43:\$R\$43;1;1+(CODE(\$C\$1)-65)*3))

Zelle **L8**: =WENN(ODER(INDEX(EURO!\$B\$36:\$R\$36;1;1+(CODE(\$C\$1)-65)*3)="";
INDEX(EURO!\$B\$36:\$R\$36;1;2+(CODE(\$C\$1)-65)*3)=""");"";
INDEX(EURO!\$B\$36:\$R\$36;1;2+(CODE(\$C\$1)-65)*3))

Zelle **L9**: =WENN(ODER(INDEX(EURO!\$B\$43:\$R\$43;1;1+(CODE(\$C\$1)-65)*3)="";
INDEX(EURO!\$B\$43:\$R\$43;1;2+(CODE(\$C\$1)-65)*3)=""");"";
INDEX(EURO!\$B\$43:\$R\$43;1;2+(CODE(\$C\$1)-65)*3))

Geschafft! - Wir haben die sechs Spiele der Gruppe A und die Spielergebnisse auf unser Tabellenblatt „GrpA“ übertragen.

Wir können die Flexibilität dieser Formeln nun einmal ausprobieren. Tragen wir z. B. in die Zelle **C1** den Buchstaben „B“ ein, sehen wir sofort die Spiele der Gruppe B mit ihren Ergebnissen!

7.2 Rangliste - Teil 1

In die Zelle **M4** tragen wir die Formel `=WENN(UND($E4<>"";$G4<>"");1;0)` ein. Hier wird geprüft, ob das Spielergebnis vollständig eingetragen wurde. Die Formel wird nach unten kopiert bis zur Zelle **M9**. In der Zelle **M10** bilden wir die Summe des Bereichs **M4:M9**. Somit haben wir die Anzahl der vollständig eingetragenen Spiele.

Zelle **H4**: `=WENN($K4>0;WENN($E4>$G4;3;WENN($E4=$G4;1;0));0)`

Zelle **I4**: `=WENN($K4>0;WENN($E4<$G4;3;WENN($E4=$G4;1;0));0)`

Die beiden Formeln werden nach unten kopiert.

Im Bereich **H4:H9** erscheinen 3 Punkte, wenn die Mannschaft 1 das Spiel gewonnen hat, bei Unentschieden 1 Punkt und 0, wenn das Spiel verloren wurde oder kein Spielergebnis eingetragen wurde. Entsprechend erscheinen im Bereich **I4:I9** die von Mannschaft 2 erzielten Punkte.

Im Bereich **O4:AF7** fertigen wir nun eine Tabelle mit den vier Mannschaften an, in der wir für jede Mannschaft eine Wertungszahl berechnen, die alle Kriterien der UEFA einbezieht. **Diese Zahl in der Spalte V wird dann über die Reihenfolge der vier Mannschaften entscheiden.**

In Spalte O steht die fortlaufende Nummer von 1 bis 4. In der Spalte P stehen die Namen der vier Mannschaften.

In Zelle **P4** tragen wir die Formel `=SVERWEIS(Q1&1;Groups!B7:D35;3;0)` ein und kopieren sie nach unten bis zur Zelle **P7**. Die rote **1** ändern wir in den Zellen **P5:P7** in **2, 3** bzw. **4**.

Der erste Parameter der Funktion SVERWEIS ist der Ausdruck `Q1 & 1`. Dieser Ausdruck wird aufgelöst zu "A1", denn in `Q1` steht ein "A". Die Funktion SVERWEIS schaut also auf dem Tabellenblatt 'Groups' nach, welcher Ländername unter "A1" zu finden ist. In Zelle **P5** wird dann nach „A2“ geschaut usw.

In der Zelle **Q4** soll die **Punktzahl** stehen, die die erste Mannschaft erreicht hat. Dazu addieren wir alle Punkte in Spalte H, wenn in Spalte B die betreffende Mannschaft steht: `SUMMEWENN(B4:B9;P4;H4:H9)`. Dazu kommen noch alle Punkte in Spalte I, wenn die betreffende Mannschaft in Spalte D steht: `SUMMEWENN(D4:D9;P4;I4:I9)`

Die Formel in **Q4** lautet deshalb:

`=SUMMEWENN(B4:B9;P4;H4:H9)+SUMMEWENN(D4:D9;P4;I4:I9)`

Die Formel kann nach unten kopiert werden bis **Q7**.

In Zelle **R4** tragen wir die Formel für die **Tordifferenz** ein: `=S4-T4`. Auch diese kann nach unten kopiert werden.

Für die **erzielten Tore** in der Spalte S und die **Gegentore** in Spalte T kommen Formeln mit demselben Aufbau wie bei der Punktzahl zum Einsatz.

Zelle **S4**: **=SUMMEWENN(\$B\$4:\$B\$9;\$P4;\$E\$4:\$E\$9)+SUMMEWENN(\$D\$4:\$D\$9;\$P4;\$G\$4:\$G\$9)**

Zelle **T4**: **=SUMMEWENN(\$B\$4:\$B\$9;\$P4;\$G\$4:\$G\$9)+SUMMEWENN(\$D\$4:\$D\$9;\$P4;\$E\$4:\$E\$9)**

In der Spalte U wird die **Anzahl der gewonnenen Spiele** ermittelt.

In Zelle **U4** tragen wir die folgende Formel ein:

=ZÄHLENWENNS(\$B\$4:\$B\$9;\$P4;\$H\$4:\$H\$9;"=3")+ZÄHLENWENNS(\$D\$4:\$D\$9;\$P4;\$I\$4:\$I\$9;"=3")

Für jedes Team wird gezählt, wie oft eine „3“ in den Spalten H und I eingetragen ist.

In den Spalten W, X, Y, Z und AA sollen die **direkten Vergleiche** berücksichtigt werden. Die Formeln dazu brauchen eine Vorbereitung und werden später besprochen (siehe Kap. 7.5 und 7.6).

Spalte AB:

In der Spalte AB sollen die **Bonuspunkte** aus dem Tabellenblatt „FairPlay“ stehen. Die Formel in Zelle **AB4** heißt:

=SUMMEWENNS(FairPlayPoints1;FairPlayTeams1;\$P4)+SUMMEWENNS(FairPlayPoints2;FairPlayTeams2;\$P4)

Die vier Variablen FairPlayPoints1, FairPlayTeams1, FairPlayPoints2 und FairPlayTeams2 legen wir im Namensmanager von Excel (Menü Formeln → Namensmanager) wie folgt fest:

FairPlayPoints1: **FairPlay!\$C\$6:\$C\$21**

FairPlayTeams1: **FairPlay!\$B\$6:\$B\$21**

FairPlayPoints2: **FairPlay!\$F\$6:\$F\$21**

FairPlayTeams2: **FairPlay!\$E\$6:\$E\$21**

In den Spalten C und F auf dem Tabellenblatt „FairPlay“ werden alle Zahlen addiert, wenn links davon der betreffende Mannschaftsname steht.

Spalte AC:

In der Spalte AC soll für jedes Team angezeigt werden, ob für das betreffende Team **schon ein Spielergebnis** eingetragen wurde (TRUE) oder noch keins (FALSE). Hat ein Team mindestens ein Spiel bestritten, können die Punktzahl und die Anzahl der Gegentore nicht beide null sein. Deshalb lautet die Formel in **AC4**:

=(Q4+T4>0)

Diese Formel wird nach unten kopiert.

Spalte AD:

In der Spalte AD berechnen wir für jedes Team eine **Ranking-Zahl**, die nur die Punkte, die Tordifferenz und die erzielten Tore berücksichtigt. Die Formel in Zelle **AD4** lautet dazu:

=\$Q4*FactorPts+(GDzero+\$R4)*FactorGD+\$S4*FactorFor

Zellen Y11 und Y12:

Um die Ergebnisse des **Elfmeterschießens** zu berücksichtigen, müssen wir zuerst prüfen, ob die Bedingungen für ein Elfmeterschießen in einem der beiden letzten Gruppenspiele überhaupt erfüllt sind. In der Zelle **Y11** soll eine „1“ stehen, wenn diese Bedingungen für das vorletzte Gruppenspiel erfüllt sind, ansonsten eine „0“. In

der Zelle **Y12** soll eine „1“ stehen, wenn diese Bedingungen für das letzte Gruppenspiel erfüllt sind, ansonsten eine „0“.

Dazu tragen wir in die Zelle **Y11** folgende Formel ein:

```
=WENN(UND(SVERWEIS($B$8;$P$4:$AD$7;15;0)=SVERWEIS($D$8;$P$4:$AD$7;15;0);$E$8=$G$8;
ZÄHLENWENN($Q$4:$Q$7;SVERWEIS($B$8;$P$4:$Q$7;2;0))=2;SUMMEWENN($B$4:$B$9;$B$8;$M$4:$M$9)
+SUMMEWENN($D$4:$D$9;$B$8;$M$4:$M$9)=3;SUMMEWENN($B$4:$B$9;$D$8;$M$4:$M$9)+
SUMMEWENN($D$4:$D$9;$D$8;$M$4:$M$9)=3);1;0)
```

Diese Formel hat folgenden Aufbau:

```
=WENN(UND(Bedingung1; Bedingung2; Bedingung3; Bedingung4; Bedingung5);1;0)
```

Die fünf Bedingungen haben folgende Bedeutung:

Bedingung1: `SVERWEIS(B8;P4:AD7;15;0)=SVERWEIS(D8;P4:AD7;15;0)`

Die beiden SVERWEIS-Funktionen schauen in Spalte 15 – das ist die Spalte AD – nach, ob die beiden Teams einen Gleichstand in Punkten, Tordifferenz und erzielten Toren haben.

Bedingung2: `E8=G8`

Das betreffende Gruppenspiel muss unentschieden enden.

Bedingung3: `ZÄHLENWENN(Q4:Q7;SVERWEIS(B8;P4:Q7;2;0))=2`

Kein weiteres Team darf dieselbe Punktzahl haben. Mit anderen Worten: Die Punktzahl, die die beiden Teams haben, muss genau zweimal vorkommen.

Die Funktion SVERWEIS ermittelt die Punktzahl eines der beiden Teams. Die Funktion ZÄHLENWENN zählt dann im Bereich \$Q\$4:\$Q\$7, wie oft diese Punktzahl vorkommt.

Bedingung4: `SUMMEWENN(B4:B9;B8;M4:M9)+SUMMEWENN(D4:D9;B8;M4:M9)=3`

Hier wird geprüft, ob für das erste Team genau drei Spiele eingetragen wurden.

Bedingung5: `SUMMEWENN(B4:B9;D8;M4:M9)+SUMMEWENN(D4:D9;D8;M4:M9)=3`

Hier wird geprüft, ob für das zweite Team genau drei Spiele eingetragen wurden.

Eine entsprechende Formel tragen wir in die Zelle **Y12** ein:

```
=WENN(UND(SVERWEIS($B$9;$P$4:$AD$7;15;0)=SVERWEIS($D$9;$P$4:$AD$7;15;0);$E$9=$G$9;
ZÄHLENWENN($Q$4:$Q$7;SVERWEIS($B$9;$P$4:$Q$7;2;0))=2;SUMMEWENN($B$4:$B$9;$B$9;$M$4:$M$9)
+SUMMEWENN($D$4:$D$9;$B$9;$M$4:$M$9)=3;SUMMEWENN($B$4:$B$9;$D$9;$M$4:$M$9)+
SUMMEWENN($D$4:$D$9;$D$9;$M$4:$M$9)=3);1;0)
```

Mit Hilfe der Ergebnisse in den Zellen **Y11** und **Y12** können wir auf dem Tabellenblatt „EURO“ eine bedingte Formatierung für die Beschriftungen „Elfmeterschießen“ der Zellen **B35** und **B42** einrichten. Wenn alle Bedingungen für ein Elfmeterschießen erfüllt sind, erscheinen diese in roter Schrift.

Spalte AE:

In die Zelle **AE4** tragen wir nun folgende Formel ein:

```
=WENN(ODER(UND($Y$11>0;ODER(UND($B$8=$P4;$J$8>$L$8);UND($D$8=$P4;$J$8<$L$8)));UND($Y$12>0;
ODER(UND($B$9=$P4;$J$9>$L$9);UND($D$9=$P4;$J$9<$L$9)));1;0)
```

In der Zelle **AE4** soll eine „1“ stehen, wenn das betreffende Team ein Elfmeterschießen gewonnen hat und alle Bedingungen für ein Elfmeterschießen erfüllt sind. Ansonsten soll in Zelle **AE4** eine „0“ stehen.

Struktur der Formel:

=WENN(ODER(Elfmeterschießen im 5. Spiel gewonnen; Elfmeterschießen im 6. Spiel gewonnen));1;0)

Der Teil „Elfmeterschießen im 5. Spiel gewonnen“ besteht aus:

UND(\$Y\$11>0;ODER(UND(\$B\$8=\$P4;\$J\$8>\$L\$8);UND(\$D\$8=\$P4;\$J\$8<\$L\$8)))

Zwei Bedingungen müssen hier erfüllt sein:

1. **\$Y\$11>0**
2. **ODER(UND(\$B\$8=\$P4;\$J\$8>\$L\$8);UND(\$D\$8=\$P4;\$J\$8<\$L\$8))**

Die **1. Bedingung (\$Y\$11>0)** besagt:

Alle Voraussetzungen für ein Elfmeterschießen im 5. Gruppenspiel sind erfüllt.

Die **2. Bedingung** besagt:

Das betreffende Team (z. B. Türkei) ist in Zelle **B8** eingetragen (**\$B\$8=\$P4**) und hat das Elfmeterschießen gewonnen (**\$J\$8>\$L\$8**)

ODER

das betreffende Team (z. B. Türkei) ist in Zelle **D8** eingetragen (**\$D\$8=\$P4**) und hat das Elfmeterschießen gewonnen (**\$J\$8<\$L\$8**).

Der Teil „Elfmeterschießen im 6. Spiel gewonnen“ ist analog aufgebaut.

Die Formel in Zelle **AE4** wird nach unten kopiert.

Spalte AF:

In der Spalte AF soll das **Ranking in der Schlussrangliste der European Qualifiers** erfasst werden. In der Spalte D auf dem Tabellenblatt ‚Language‘ stehen die Rangzahlen für jedes Team. Dass diese Rangzahlen dort in aufsteigender Reihenfolge stehen, ist nur Kosmetik. Sie dürfen auch anders angeordnet sein, da sie mit Hilfe der Funktionen INDEX und VERGLEICH gesucht werden.

In die Zelle AF4 tragen wir folgende Formel ein:

=INDEX(Language!\$D\$5:\$D\$59;VERGLEICH(\$P4;Language!\$E\$5:\$E\$59;0);1)

Der Ausdruck **VERGLEICH(\$P4;Language!\$E\$5:\$E\$59;0)** liefert die relative Position des Teams in der Liste aller Teams (Spalte E). Die Funktion INDEX gibt dann die zugehörige Rangzahl zurück.

7.3 Das Tabellenblatt Factors

Das Sortieren der vier Mannschaften in die richtige Rangreihenfolge geschieht nach folgendem Prinzip:

In der Spalte V des Tabellenblattes „GrpA“ steht für jede Mannschaft eine „**Rangzahl**“, die über den Rang entscheidet. Die Mannschaft mit der größten Zahl steht an erster Stelle.

Diese Zahl entsteht auf folgende Weise:

Höchste Priorität hat die erreichte Punktzahl in Spalte P. Diese Punktzahl wird deshalb mit einem sehr großen Faktor multipliziert. Wir nehmen die Zahl 10000000. Zweithöchste Priorität hat der direkte Vergleich dreier Mannschaften (siehe Kap. 7.5) in Spalte W. Er wird mit dem niedrigeren Faktor 1000000 multipliziert und hinzuaddiert. Dritthöchste Priorität hat der direkte Vergleich zweier Mannschaften in Spalte X. Er wird mit dem Faktor 100000 multipliziert und ebenfalls hinzuaddiert. Die Faktoren für die restlichen Kriterien (Tordifferenz,

erzielte Tore usw.) haben kleinere Faktoren: 1000, 10, usw.

Diese Faktoren hinterlegen wir auf einem extra Tabellenblatt mit dem Namen „**Factors**“. Hier gibt es keine Formeln, nur Texte und Zahlen. Es sieht so aus:

	A	B	C	D	E	F
1		Factors				
2						
3		Factors	Category			
4		10000000	Points			
5		1000000	direct Comp. 3			
6		100000	direct Comp. 2			
7		1000	Goal difference	50	Goal difference zero	
8		10	Goals scored			
9		1	Penalty shootout			
10		0,1	direct Comp. 4.3			
11		0,01	direct Comp. 4.2			
12		0,001	Wins			
13		0,0001	Fair play			
14		0,000001	Ranking			
15		0,0000001	Row			
16						

Die einzelnen Faktoren in den Zellen **B4** bis **B15** erhalten im Namensmanager folgende Variablennamen:

B4:	FactorPts	(Faktor Punkte)
B5:	FactorDirC3	(Faktor Direktvergleiche dreier Teams)
B6:	FactorDirC2	(Faktor Direktvergleiche zweier Teams)
B7:	FactorGD	(Faktor Tordifferenz)
B8:	FactorFor	(Faktor erzielte Tore)
B9:	FactorPenalty	(Faktor Elfmeterschießen)
B10:	FactorDirC43	(Faktor Direktvergleiche dreier Teams bei Punktgleichheit aller vier Teams)
B11:	FactorDirC42	(Faktor Direktvergleiche zweier Teams bei Punktgleichheit aller vier Teams)
B12:	FactorWins	(Faktor Anzahl der Siege)
B13:	FactorFairPlay	(Faktor Fairplay)
B14:	FactorRank	(Faktor Rangliste)
B15:	FactorRow	(Faktor Zeile)

Somit können wir in den Formeln an Stelle von **Factors!\$B\$4** einfach **FactorPts** schreiben. Die Formeln werden dann besser lesbar.

7.4 Rangliste - Teil 2

Die Formeln in **W4**, **X4**, **Y4**, **Z4**, und **AA4** werden später besprochen (siehe Kap. 7.5 und 7.6).

In Zelle **V4** tragen wir die folgende Formel ein:

```
= $Q4*FactorPts+(GDzero+$R4)*FactorGD+$S4*FactorFor+$U4*FactorWins+$W4*FactorDirC3+$X4*FactorDirC2+$Y4*FactorDirC43+$Z4*FactorDirC42+$AA4*FactorDirC42+$AB4*FactorFairPlay+$AE4*FactorPenalty+(100-$AF4)*($M$10>0)*FactorRank
```

GDzero wird zur Tordifferenz **\$R4** addiert, damit dieser Wert nicht negativ werden kann. Diese Formel ist noch unvollständig und wird noch erweitert werden.

Damit haben wir eine komplexe Rangzahl, die alle Kriterien der UEFA mit der richtigen Priorität berücksichtigt.

Erklärung des letzten Ausdrucks: **$(100 - \text{SAF4}) * (\text{M10} > 0) * \text{FactorRank}$**

Die Zahl **$100 - \text{SAF4}$** wird umso größer, je kleiner die Rangzahl **SAF4** des Teams ist. Das Team mit der Rangzahl 1 hat den größtmöglichen Faktor 99. Der Faktor **$(\text{M10} > 0)$** wird 0 (= FALSE), wenn **$\text{M10} = 0$** ist, wenn also noch kein Spielergebnis eingetragen wurde. In diesem Fall sollen die Teams in ihrer ursprünglichen Reihenfolge erscheinen und nicht in der Reihenfolge des Qualifiers-Rankings.

Anhand der berechneten Rangzahlen im Bereich **$\text{V4}:\text{V7}$** können wir nun die Gruppentabelle sortieren. Die Mannschaft mit der höchsten Rangzahl soll an erster Stelle stehen. Im Bereich **$\text{O11}:\text{V14}$** werden wir eine **sortierte Liste der vier Mannschaften** erzeugen.

In die Zelle **V11** tragen wir folgende Formel ein: **$=\text{KGRÖSSTE}(\text{\$V\$4}:\text{\$V\$7};\text{ZEILE}(\text{A1}))$** und kopieren sie nach unten bis **V14** .

Die Formel **$=\text{KGRÖSSTE}(\text{Bereich};1)$** gibt den größten Wert von Bereich zurück,
 die Formel **$=\text{KGRÖSSTE}(\text{Bereich};2)$** den zweitgrößten,
 die Formel **$=\text{KGRÖSSTE}(\text{Bereich};3)$** den drittgrößten,
 die Formel **$=\text{KGRÖSSTE}(\text{Bereich};4)$** den viertgrößten.

Um die erste Formel in **V14** einfach nach unten kopieren zu können, erzeugt man die Zahlen 1, 2, 3 und 4 automatisch. Der Ausdruck **$\text{ZEILE}(\text{A1})$** gibt die Zeilennummer von **A1** zurück, also 1. Beim Kopieren nach unten wird daraus **$\text{ZEILE}(\text{A2})$** , **$\text{ZEILE}(\text{A3})$** , **$\text{ZEILE}(\text{A4})$** , es werden also die Zahlen 2, 3 und 4 erzeugt.

Im Bereich **$\text{O11}:\text{V14}$** wollen wir nun die gleiche Gruppentabelle wie im Bereich **$\text{O4}:\text{V7}$** erzeugen, nur diesmal sortiert. Dazu muss als erstes die Rangzahl der Zelle **V11** oben im Bereich **$\text{V4}:\text{V7}$** gesucht werden. Wird sie zum Beispiel in der Zelle **V5** gefunden, werden alle Werte links von Zelle **V5** (die Werte der Zellen **$\text{P5}:\text{U5}$**) nach unten in die Zellen **$\text{P11}:\text{U11}$** übertragen.

Die Funktion **SVRWEIS** ist hier nicht geeignet, weil sie nur Werte rechts von der Suchspalte finden kann. In diesem Fall hilft eine Kombination der Funktionen **INDEX** und **VERGLEICH** .

Zunächst tragen wir in die Zellen **O11** bis **O14** wieder die laufende Nummer ein. In die Zelle **P11** tragen wir die folgende Formel ein:

$=\text{INDEX}(\text{P\$4}:\text{P\$7};\text{VERGLEICH}(\text{\$V11};\text{\$V\$4}:\text{\$V\$7};0))$

Das Dollarzeichen vor dem „P“ fehlt, weil die Formel nach rechts kopiert werden soll. Dann wird aus „P“ ein „Q“ usw.

Erklärung der Formel:

Die Formel **$=\text{INDEX}(\text{P\$4}:\text{P\$7};3)$** würde aus dem Bereich **$\text{P\$4}:\text{P\$7}$** den dritten Wert zurückgeben, also den Inhalt der Zelle **P6** . Die "3" wird nun durch den Ausdruck **$\text{VERGLEICH}(\text{\$V11};\text{\$V\$4}:\text{\$V\$7};0)$** ersetzt. Die Funktion **VERGLEICH** liefert die richtige Zahl, weil sie den Inhalt von **$\text{\$V11}$** im Bereich **$\text{\$V\$4}:\text{\$V\$7}$** sucht und die Position des gefundenen Wertes zurückgibt. Der dritte Parameter der Funktion **VERGLEICH** ist eine Null. Das bedeutet, dass die Funktion nach einer genauen Übereinstimmung sucht.

Die Formel in **P11** wird nach rechts und nach unten kopiert bis **U14** und schon ist die sortierte Gruppentabelle fertig!

Einen Fehler hat die sortierte Gruppentabelle allerdings noch:

Wenn z. B. die zweitplatzierte und die drittplatzierte Mannschaft dieselbe Rangzahl haben, findet die Funktion VERGLEICH nur noch die Position der zweiten Mannschaft, aber nicht mehr die Position der dritten Mannschaft. In der Tabelle steht dann zweimal der Name der zweiten Mannschaft mit ihren Werten, die dritte Mannschaft fehlt. Wir müssen also vermeiden, dass zwei oder mehr Rangzahlen gleich sind. Sind z. B. alle vier gleich, weil noch keine Spielergebnisse eingetragen wurden, erscheint in der sortierten Liste viermal die erste Mannschaft. Statt dessen sollen aber die vier Mannschaften in ihrer ursprünglichen Reihenfolge erscheinen. Dies erreichen wir, indem wir zur Rangzahl noch die Zeilennummer multipliziert mit dem Faktor FactorRow (=0,00000001) addieren.

Die endgültige Formel in Zelle **V4** heißt dann:

```
= $Q4*FactorPts+(GDzero+$R4)*FactorGD+$S4*FactorFor+$U4*FactorWins+$W4*FactorDirC3+$X4*FactorDirC2+$Y4*FactorDirC43+$Z4*FactorDirC42+$AA4*FactorDirC42+$AB4*FactorFairPlay+$AE4*FactorPenalty+(100-$AF4)*($M$10>0)*FactorRank+(8-ZEILE())*FactorRow
```

Die Funktion **ZEILE()** gibt die Zeilennummer der Zelle zurück, in der die Formel steht.

Der Ausdruck **8-ZEILE()** erzeugt für **V4** die Zahl 4, für **V5** die Zahl 3, für **V6** die Zahl 2 und für **V7** die Zahl 1.

Somit erscheint die Mannschaft in Zeile 4 bei Gleichheit aller Mannschaften an oberster Stelle, die Mannschaft in Zeile 5 an zweiter Stelle usw.

7.5 Direkter Vergleich dreier Mannschaften

Der direkte Vergleich von zwei oder drei Mannschaften kommt im Gegensatz zur Fußballweltmeisterschaft der FIFA viel häufiger vor. Das liegt daran, dass dieser Fall bei der Europameisterschaft bereits bei Punktgleichheit der betreffenden Mannschaften eintritt. Bei der FIFA Weltmeisterschaft dagegen wird ein direkter Vergleich nur dann durchgeführt, wenn die betreffenden Mannschaften die gleiche Punktzahl, die gleiche Tordifferenz und die gleiche Anzahl erzielter Tore haben. Dies ist seltener, kommt aber trotzdem gelegentlich vor.

Bei der Europameisterschaft der UEFA spielen die direkten Vergleiche auch deshalb eine so große Rolle, weil sie Vorrang haben vor der Tordifferenz und der Anzahl erzielter Tore.

Um einen direkten Vergleich dreier Mannschaften durchzuführen, legen wir vier Gruppentabellen mit jeweils drei Mannschaften an, in denen nur die Spiele berücksichtigt werden, die diese drei Mannschaften gegeneinander bestritten haben.

		Points	GD	For	Against		Pts/GD/For	Rank Col. V
1.	Türkei	0	-5	0	5		45.000	1
2.	Italien	6	4	4	0		60.054.040	3
3.	Wales	3	1	2	1		30.051.020	2
	Schweiz							

Für die Mannschaftsnamen tragen wir die folgenden Formeln ein:

Zelle **P18**: **= \$P\$4**

Zelle **P19**: **= \$P\$5**

Zelle **P20**: **= \$P\$6**

Zelle **P21**: **= \$P\$7**

In Zelle **Q18** tragen wir folgende Formel ein:

```
=SUMMEWENNS($H$4:$H$9;$B$4:$B$9;$P18;$D$4:$D$9;"<>"&$P$21)
+SUMMEWENNS($I$4:$I$9;$B$4:$B$9;"<>"&$P$21;$D$4:$D$9;$P18)
```

Parameter der Funktion SUMMEWENNS:

\$H\$4:\$H\$9 **Summen-Bereich:** der Bereich, in dem die Zahlen unter bestimmten Bedingungen addiert werden

\$B\$4:\$B\$9 **Kriterien-Bereich 1:** der Bereich, der auf das Kriterium 1 geprüft wird

\$P18 **Kriterium 1:** der Inhalt von **\$P18** wird im Kriterien-Bereich 1 auf Übereinstimmung geprüft

\$D\$4:\$D\$9 **Kriterien-Bereich 2:** der Bereich, der auf das Kriterium 2 geprüft wird

"<>"&\$P\$21 **Kriterium 2:** im Kriterien-Bereich 2 wird geprüft, ob der Inhalt ungleich dem Inhalt von **\$P\$21** ist

Die zweite Funktion SUMMEWENNS in der Formel hat denselben Aufbau. Für jede der drei Mannschaften werden die erreichten Punkte addiert, wenn sie nicht gegen die vierte Mannschaft in der Zelle **P21** erzielt wurden.

Die Formeln in **S18** und **T18** arbeiten nach demselben Prinzip:

Zelle **S18:**

```
=SUMMEWENNS($E$4:$E$9;$B$4:$B$9;$P18;$D$4:$D$9;"<>"&$P$21)
+SUMMEWENNS($G$4:$G$9;$B$4:$B$9;"<>"&$P$21;$D$4:$D$9;$P18)
```

Zelle **T18:**

```
=SUMMEWENNS($G$4:$G$9;$B$4:$B$9;$P18;$D$4:$D$9;"<>"&$P$21)
+SUMMEWENNS($E$4:$E$9;$B$4:$B$9;"<>"&$P$21;$D$4:$D$9;$P18)
```

Zelle **R18:** =S18-T18

Zelle **V18:** =Q18*FactorPts+(GDzero+R18)*FactorGD+S18*FactorFor

Beim direkten Vergleich werden die erreichten Punkte, die Tordifferenz und die erzielten Tore miteinander verglichen. In der Zelle **V18** steht also eine Bewertungszahl für den direkten Vergleich.

Die Formeln von **Q18** bis **V18** werden nun nach unten kopiert bis zur Zeile 20.

In der Spalte W erzeugen wir nun eine Rangnummer für jede der drei Mannschaften.

Zelle **W18:** =RANG(V18;V\$18:V\$20;1)

Die Funktion RANG ermittelt hier den Rang der Zahl in Zelle **V18** innerhalb des Bereiches **V\$18:V\$20**. Der dritte Parameter, die "1", bewirkt, dass die größte Zahl den niedrigsten Rang, also Rang 3, bekommt.

Die Rangzahl in Zelle **W18** wird nun nach Zelle **X18** übertragen, wenn diese Mannschaft punktgleich ist mit der zweiten und dritten Mannschaft, aber nicht mit der vierten. Andernfalls wird in Zelle **X18** eine Null eingetragen. Damit ist gewährleistet, dass die in Zelle **W18** ermittelte Rangzahl nur berücksichtigt wird, wenn die Voraussetzung für einen direkten Vergleich erfüllt ist.

Zelle **X18:** =WENN(UND(\$Q\$4=\$Q\$5;\$Q\$4=\$Q\$6;\$Q\$4<>\$Q\$7);W18;0)

In der Zelle **W4** werden alle Zahlen des Bereichs **X18:X35** addiert, wenn in der P-Spalte der Mannschaftsname ‚Türkei‘ steht. Dies bewerkstelligt die folgende Formel:

Zelle **W4:** =SUMMEWENNS(\$X\$18:\$X\$35;\$P\$18:\$P\$35;\$P4)

Da jede Mannschaft im Bereich **X18:X35** nur einmal an einem direkten Vergleich dreier Mannschaften beteiligt

sein kann, steht auf diese Weise in der Zelle **W4** die Rangzahl des direkten Vergleichs oder eine Null. Die Formel in Zelle **W4** wird nach unten kopiert bis Zelle **W7**.

Wir markieren den Bereich Q18:X18 und kopieren die Formeln nach unten in die Zeilen 19 und 20.

Die restlichen Formeln im Bereich **P18:X35** sehen ganz ähnlich aus:

Gruppe 2

Zelle **P23**: =SP\$4

Zelle **P24**: =SP\$5

Zelle **P25**: =SP\$7

Zelle **P26**: =SP\$6

Zelle **Q23**:

=SUMMEWENNS(\$H\$4:\$H\$9;\$B\$4:\$B\$9;\$P23;\$D\$4:\$D\$9;"<>"&SP\$26)
+SUMMEWENNS(\$I\$4:\$I\$9;\$B\$4:\$B\$9;"<>"&SP\$26;\$D\$4:\$D\$9;\$P23)

Zelle **R23**: =S23-T23

Zelle **S23**:

=SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P23;\$D\$4:\$D\$9;"<>"&SP\$26)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;"<>"&SP\$26;\$D\$4:\$D\$9;\$P23)

Zelle **T23**:

=SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P23;\$D\$4:\$D\$9;"<>"&SP\$26)
+SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;"<>"&SP\$26;\$D\$4:\$D\$9;\$P23)

Zelle **V23**: =Q23*FactorPts+(GDzero+R23)*FactorGD+S23*FactorFor

Zelle **W23**: =RANG(V23;V\$23:V\$25;1)

Zelle **X23**: =WENN(UND(\$Q\$4=\$Q\$5;\$Q\$4=\$Q\$7;\$Q\$4<>\$Q\$6);W23;0)

Gruppe 3

Zelle **P28**: =SP\$4

Zelle **P29**: =SP\$6

Zelle **P30**: =SP\$7

Zelle **P31**: =SP\$5

Zelle **Q28**:

=SUMMEWENNS(\$H\$4:\$H\$9;\$B\$4:\$B\$9;\$P28;\$D\$4:\$D\$9;"<>"&SP\$31)
+SUMMEWENNS(\$I\$4:\$I\$9;\$B\$4:\$B\$9;"<>"&SP\$31;\$D\$4:\$D\$9;\$P28)

Zelle **R28**: =S28-T28

Zelle **S28**:

=SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P28;\$D\$4:\$D\$9;"<>"&SP\$31)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;"<>"&SP\$31;\$D\$4:\$D\$9;\$P28)

Zelle **T28**:

```
=SUMMEWENNS($G$4:$G$9;$B$4:$B$9;$P28;$D$4:$D$9;"<>"&$P$31)
+SUMMEWENNS($E$4:$E$9;$B$4:$B$9;"<>"&$P$31;$D$4:$D$9;$P28)
```

Zelle **V28**: =Q28*FactorPts+(GDzero+R28)*FactorGD+S28*FactorFor

Zelle **W28**: =RANG.GLEICH(V28;V\$28:V\$30;1)

Zelle **X28**: =WENN(UND(\$Q\$4=\$Q\$6;\$Q\$4=\$Q\$7;\$Q\$4<>\$Q\$5);W28;0)

Gruppe 4

Zelle **P33**: =\$P\$5

Zelle **P34**: =\$P\$6

Zelle **P35**: =\$P\$7

Zelle **P36**: =\$P\$4

Zelle **Q33**:

```
=SUMMEWENNS($H$4:$H$9;$B$4:$B$9;$P33;$D$4:$D$9;"<>"&$P$36)
+SUMMEWENNS($I$4:$I$9;$B$4:$B$9;"<>"&$P$36;$D$4:$D$9;$P33)
```

Zelle **R33**: =S33-T33

Zelle **S33**:

```
=SUMMEWENNS($E$4:$E$9;$B$4:$B$9;$P33;$D$4:$D$9;"<>"&$P$36)
+SUMMEWENNS($G$4:$G$9;$B$4:$B$9;"<>"&$P$36;$D$4:$D$9;$P33)
```

Zelle **T33**:

```
=SUMMEWENNS($G$4:$G$9;$B$4:$B$9;$P33;$D$4:$D$9;"<>"&$P$36)
+SUMMEWENNS($E$4:$E$9;$B$4:$B$9;"<>"&$P$36;$D$4:$D$9;$P33)
```

Zelle **V33**: =Q33*FactorPts+(GDzero+R33)*FactorGD+S33*FactorFor

Zelle **W33**: =RANG(V33;V\$33:V\$35;1)

Zelle **X33**: =WENN(UND(\$Q\$5=\$Q\$6;\$Q\$5=\$Q\$7;\$Q\$5<>\$Q\$4);W33;0)

In die Zelle **W4** tragen wir folgende Formel ein:

```
=SUMMEWENNS($X$18:$X$35;$P$18:$P$35;$P4)
```

Das bedeutet:

Im Bereich **X18:X35** werden alle Zahlen addiert, die zur ersten Mannschaft (**\$P4**) gehören. Wenn also ein direkter Vergleich für diese Mannschaft stattfindet, dann erscheint in der Zelle **W4** eine Zahl größer als Null. Diese Zahl wird mit dem Faktor **FactorDirC3** multipliziert und in der Formel für die Rangzahl (in Zelle **V4**) berücksichtigt (siehe Kap. 7.4).

Die Formel in **W4** kopieren wir nach unten bis **W7**.

Damit ist der direkte Vergleich von drei Teams noch nicht abgeschlossen, denn es kann vorkommen, dass zwei der drei Mannschaften immer noch nicht unterscheidbar sind. In diesem Fall findet ein weiterer direkter

Vergleich dieser beiden Mannschaften statt.

Spezialfall:

Zusätzlicher Direktvergleich zweier Mannschaften

Um die Formeln für diesen Fall zu vereinfachen, kopieren wir in den Bereich **AA18:AC35** die Spielergebnisse der direkten Begegnungen. Zum Beispiel findet man in den Zellen **AA18** und **AA19** das Spielergebnis des Spiels Türkei-Italien, in den Zellen **AB18** und **AB20** das Spielergebnis des Spiels Türkei-Wales usw.

Die Spielergebnisse werden aus dem Bereich **E4:G9** kopiert. In Zelle **AA18** steht die Formel **=SI\$18**, in Zelle **AA19** die Formel **=H\$19** usw.

In Zelle **Y18** tragen wir folgende Formel ein:

```
=WENN(UND(V18=V19;V18<>V20);WENN(AA18>AA19;3;WENN(AA18=AA19;1;0));WENN(UND(V18=V20;V18<>V19);WENN(AB18>AB20;3;WENN(AB18=AB20;1;0));""))
```

In der äußeren WENN-Funktion **WENN(UND(V18=V19;V18<>V20);<Ja>;<Nein>)** wird zuerst geprüft, ob die Mannschaft in Zeile 18 (Türkei) punktgleich ist mit der Mannschaft in Zeile 19 (Italien), aber nicht punktgleich ist mit der Mannschaft in Zeile 20 (Wales).

Falls **<Ja>**:

Der Ausdruck **WENN(AA18>AA19;3;WENN(AA18=AA19;1;0))** bedeutet, dass je nach Spielergebnis die Punktzahl 3, 1 oder 0 eingetragen wird.

Falls **<Nein>**:

Der Ausdruck **WENN(UND(V18=V20;V18<>V19);WENN(AB18>AB20;3;WENN(AB18=AB20;1;0));"")** prüft, ob die Mannschaft in Zeile 18 (Türkei) punktgleich ist mit der Mannschaft in Zeile 20 (Wales), aber nicht punktgleich mit der Mannschaft in Zeile 19 (Italien) ist. Wenn ja, wird je nach Spielergebnis die Punktzahl 3, 1, oder 0 eingetragen, wenn nein, ist die Zelle leer.

Zusammenfassung:

Wenn ein direkter Vergleich der Mannschaft in Zeile 18 mit einer der beiden anderen stattfindet, steht in Zelle **Y18** die Punktzahl der direkten Begegnung (3, 1 oder 0).

Die Formeln in den restlichen Zellen der Spalte Y lauten entsprechend:

Zelle **Y19**:

```
=WENN(UND(V19=V18;V19<>V20);WENN(AA19>AA18;3;WENN(AA19=AA18;1;0));WENN(UND(V19=V20;V19<>V18);WENN(AC19>AC20;3;WENN(AC19=AC20;1;0));""))
```

Zelle **Y20**:

```
=WENN(UND(V20=V18;V20<>V19);WENN(AB20>AB18;3;WENN(AB20=AB18;1;0));WENN(UND(V20=V19;V20<>V18);WENN(AC20>AC19;3;WENN(AC20=AC19;1;0));""))
```

Zelle **Y23**:

```
=WENN(UND(V23=V24;V23<>V25);WENN(AA23>AA24;3;WENN(AA23=AA24;1;0));WENN(UND(V23=V25;V23<>V24);WENN(AB23>AB25;3;WENN(AB23=AB25;1;0));""))
```

Zelle **Y24**:

```
=WENN(UND(V24=V23;V24<>V25);WENN(AA24>AA23;3;WENN(AA24=AA23;1;0));WENN(UND(V24=V25;V24<>V23);WENN(AC24>AC25;3;WENN(AC24=AC25;1;0));""))
```

Zelle **Y25**:

```
=WENN(UND(V25=V23;V25<>V24);WENN(AB25>AB23;3;WENN(AB25=AB23;1;0));WENN(UND(V25=V24;V25<>V23);WENN(AC25>AC24;3;WENN(AC25=AC24;1;0));""))
```

Zelle **Y28**:

```
=WENN(UND(V28=V29;V28<>V30);WENN(AA28>AA29;3;WENN(AA28=AA29;1;0));WENN(UND(V28=V30;V28<>V29);WENN(AB28>AB30;3;WENN(AB28=AB30;1;0));""))
```

Zelle **Y29**:

```
=WENN(UND(V29=V28;V29<>V30);WENN(AA29>AA28;3;WENN(AA29=AA28;1;0));WENN(UND(V29=V30;V29<>V28);WENN(AC29>AC30;3;WENN(AC29=AC30;1;0));""))
```

Zelle **Y30**:

```
=WENN(UND(V30=V28;V30<>V29);WENN(AB30>AB28;3;WENN(AB30=AB28;1;0));WENN(UND(V30=V29;V30<>V28);WENN(AC30>AC29;3;WENN(AC30=AC29;1;0));""))
```

Zelle **Y33**:

```
=WENN(UND(V33=V34;V33<>V35);WENN(AA33>AA34;3;WENN(AA33=AA34;1;0));WENN(UND(V33=V35;V33<>V34);WENN(AB33>AB35;3;WENN(AB33=AB35;1;0));""))
```

Zelle **Y34**:

```
=WENN(UND(V34=V33;V34<>V35);WENN(AA34>AA33;3;WENN(AA34=AA33;1;0));WENN(UND(V34=V35;V34<>V33);WENN(AC34>AC35;3;WENN(AC34=AC35;1;0));""))
```

Zelle **Y35**:

```
=WENN(UND(V35=V33;V35<>V34);WENN(AB35>AB33;3;WENN(AB35=AB33;1;0));WENN(UND(V35=V34;V35<>V33);WENN(AC35>AC34;3;WENN(AC35=AC34;1;0));""))
```

Die Punktzahlen in Spalte Y sollen aber nur berücksichtigt werden, wenn bereits ein Direktvergleich der drei Mannschaften vorliegt, wenn also in der Spalte X ein Wert größer als null steht. Deshalb tragen wir in Zelle **Z18** die Formel **=WENN(X18>0;Y18;"")** ein und kopieren sie nach unten bis zur Zelle **Z35**.

In der Spalte Z stehen also die Punktzahlen des Direktvergleichs der betreffenden Mannschaft mit einer anderen, wenn die betreffende Mannschaft gleichzeitig an einem Direktvergleich dreier Mannschaften beteiligt ist.

Die Ergebnisse im Bereich **Z18:Z35** werden in den Formeln des Bereichs **X4:X7** berücksichtigt (siehe Kap. 7.6). Die Formeln im Bereich **X4:X7** berücksichtigen zusätzlich noch die Direktvergleiche zweier Mannschaften in den Fällen, wo kein Direktvergleich dreier Mannschaften stattfindet. Und diese Fälle betrachten wir im nächsten Kapitel.

7.6 Direkter Vergleich zweier Mannschaften

Um den direkten Vergleich zweier Mannschaften zu berücksichtigen, legen wir zuerst im Bereich **E17:K21** eine kleine Tabelle mit den Ergebnissen der direkten Begegnungen an.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1		Group A													
3			Match		Score		Points		Penalties shoot		valid				
4		Türkei	- Italien		0 - 3		0 3					1		1.	Türkei
5		Wales	- Schweiz		1 - 1		1 1					1		2.	Italien
6		Türkei	- Wales		0 - 2		0 3					1		3.	Wales
7		Italien	- Schweiz		1 - 0		3 0					1		4.	Schweiz
8		Schweiz	- Türkei		3 - 1		3 0		:			1			
9		Italien	- Wales		1 - 0		3 0		:			1			
10									Sum:		6				Sorted
11														1.	Italien
12														2.	Schweiz
13														3.	Wales
14														4.	Türkei
16															
17															
18															
19															
20															
21															
22															
23															

Direct Comparison of 2						
	Tür	Ita	Wal	Sch	DC(2)	DC(4,2)
Türkei		0	0	1		
Italien	3		1	1		
Wales	2	0		1	1	
Schweiz	3	0	1		1	

In die Zellen **E18** bis **E21** übertragen wir die Mannschaftsnamen mit den Formeln **=SP\$4**, **=SP\$5**, **=SP\$6** und **=SP\$7**. Wir verbinden jeweils die Zellen **E18:G18**, **E19:G19**, **E20:G20** und **E21:G21**. Hier müssen die vollständigen Namen stehen! Für die Beschriftung oben in den Zellen **H17** bis **K17** genügen die ersten drei Buchstaben der Mannschaftsnamen:

Zelle **H17**: **=LINKS(SP\$4;3)**

Entsprechend in den anderen Zellen.

In der Zelle **I18** soll die Anzahl der Tore stehen, die Mannschaft 1 gegen Mannschaft 2 erzielt hat. In Zelle **H19** steht dann die Anzahl der Tore, die Mannschaft 2 gegen Mannschaft 1 erzielt hat. Wir tragen in **I18** folgende Formel ein:

**=SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$4;\$D\$4:\$D\$9;\$P\$5)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$5;\$D\$4:\$D\$9;\$P\$4)**

Im Bereich **\$E\$4:\$E\$9** werden die Zahlen addiert, wenn links die Mannschaft 1 steht und rechts die Mannschaft 2.

Im Bereich **\$G\$4:\$G\$9** werden die Zahlen addiert, wenn links die Mannschaft 2 steht und rechts die Mannschaft 1.

Auf diese Weise erhalten wir genau die Anzahl der Tore, die Mannschaft 1 gegen Mannschaft 2 erzielt hat.

Die restlichen Formeln arbeiten auf dieselbe Weise. Sie lauten:

Zelle **G18**: **=SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$4;\$D\$4:\$D\$9;\$P\$6)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$6;\$D\$4:\$D\$9;\$P\$4)**

Zelle **H18**: **=SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$4;\$D\$4:\$D\$9;\$P\$7)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$7;\$D\$4:\$D\$9;\$P\$4)**

Zelle **E19**: =SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$5;\$D\$4:\$D\$9;\$P\$4)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$4;\$D\$4:\$D\$9;\$P\$5)

Zelle **G19**: =SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$5;\$D\$4:\$D\$9;\$P\$6)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$6;\$D\$4:\$D\$9;\$P\$5)

Zelle **H19**: =SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$5;\$D\$4:\$D\$9;\$P\$7)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$7;\$D\$4:\$D\$9;\$P\$5)

Zelle **E20**: =SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$6;\$D\$4:\$D\$9;\$P\$4)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$4;\$D\$4:\$D\$9;\$P\$6)

Zelle **F20**: =SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$6;\$D\$4:\$D\$9;\$P\$5)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$5;\$D\$4:\$D\$9;\$P\$6)

Zelle **H20**: =SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$6;\$D\$4:\$D\$9;\$P\$7)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$7;\$D\$4:\$D\$9;\$P\$6)

Zelle **E21**: =SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$7;\$D\$4:\$D\$9;\$P\$4)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$4;\$D\$4:\$D\$9;\$P\$7)

Zelle **F21**: =SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$7;\$D\$4:\$D\$9;\$P\$5)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$5;\$D\$4:\$D\$9;\$P\$7)

Zelle **G21**: =SUMMEWENNS(\$E\$4:\$E\$9;\$B\$4:\$B\$9;\$P\$7;\$D\$4:\$D\$9;\$P\$6)
+SUMMEWENNS(\$G\$4:\$G\$9;\$B\$4:\$B\$9;\$P\$6;\$D\$4:\$D\$9;\$P\$7)

Ein direkter Vergleich zweier Mannschaften findet statt, wenn beide Mannschaften die gleiche Punktzahl haben.

Im Bereich **L18:L21** bekommt nun eine Mannschaft eine 3 eingetragen, wenn sie einen direkten Vergleich gewonnen hat. Sie bekommt eine 1 eingetragen, wenn sie im direkten Vergleich unentschieden gespielt hat, und eine 0, wenn sie im direkten Vergleich verloren hat. Ansonsten bleiben diese Zellen leer. Dazu tragen wir in die Zelle **L18** folgende Formel ein:

```
=WENN(UND($Q$4=$Q$5;$Q$4<>$Q$6;$Q$4<>$Q$7);WENN(I18>H19;3;WENN(I18=H19;1;0));  
WENN(UND($Q$4=$Q$6;$Q$4<>$Q$5;$Q$4<>$Q$7);WENN(J18>H20;3;WENN(J18=H20;1;0));  
WENN(UND($Q$4=$Q$7;$Q$4<>$Q$5;$Q$4<>$Q$6);WENN(K18>H21;3;WENN(K18=H21;1;0));""))
```

Die erste WENN-Funktion bedeutet: Wenn Mannschaft 1 mit Mannschaft 2 Gleichstand hat in Punkten, aber ansonsten mit keiner anderen Mannschaft ($\$Q\$4=\$Q\$5;\$Q\$4<>\$Q\$6;\$Q\$4<>\$Q\7), dann bekommt sie für einen Sieg gegen Mannschaft 2 ($I18>H19$) 3 Punkte, für ein Unentschieden ($I18=H19$) 1 Punkt und für eine Niederlage 0 Punkte.

Die zweite und die dritte WENN-Funktion testen, ob ein gewonnener direkter Vergleich von Mannschaft 1 mit Mannschaft 3 oder Mannschaft 4 vorliegt.

Die Formeln für die Zellen **L19**, **L20** und **L21** lauten entsprechend:

Zelle **L19**:

```
=WENN(UND($Q$5=$Q$4;$Q$5<>$Q$6;$Q$5<>$Q$7);WENN(H19>I18;3;WENN(H19=I18;1;0));
WENN(UND($Q$5=$Q$6;$Q$5<>$Q$4;$Q$5<>$Q$7);WENN(J19>I20;3;WENN(J19=I20;1;0));
WENN(UND($Q$5=$Q$7;$Q$5<>$Q$4;$Q$5<>$Q$6);WENN(K19>I21;3;WENN(K19=I21;1;0));""))
```

Zelle **L20**:

```
=WENN(UND($Q$6=$Q$4;$Q$6<>$Q$5;$Q$6<>$Q$7);WENN(H20>J18;3;WENN(H20=J18;1;0));
WENN(UND($Q$6=$Q$5;$Q$6<>$Q$4;$Q$6<>$Q$7);WENN(I20>J19;3;WENN(I20=J19;1;0));
WENN(UND($Q$6=$Q$7;$Q$6<>$Q$4;$Q$6<>$Q$5);WENN(K20>J21;3;WENN(K20=J21;1;0));""))
```

Zelle **L21**:

```
=WENN(UND($Q$7=$Q$4;$Q$7<>$Q$5;$Q$7<>$Q$6);WENN(H21>K18;3;WENN(H21=K18;1;0));
WENN(UND($Q$7=$Q$5;$Q$7<>$Q$4;$Q$7<>$Q$6);WENN(I21>K19;3;WENN(I21=K19;1;0));
WENN(UND($Q$7=$Q$6;$Q$7<>$Q$4;$Q$7<>$Q$5);WENN(J21>K20;3;WENN(J21=K20;1;0));""))
```

In der Zelle **X4** geben wir die Formel

```
=WENN($L18="";0;$L18)+SUMMEWENNS($Z$18:$Z$35;$P$18:$P$35;$P4)
```

ein und kopieren diese Formel nach unten bis **X7**.

Der erste Summand **WENN(\$L18="";0;\$L18)** enthält ggf. die Punktzahl aus einem Direktvergleich zweier Mannschaften, der zweite Summand **SUMMEWENNS(\$Z\$18:\$Z\$35;\$P\$18:\$P\$35;\$P4)** die Punktzahl aus dem Direktvergleich mit einer anderen Mannschaft, wenn gleichzeitig ein Direktvergleich dreier Mannschaften stattfindet. (Es kann immer nur einer der beiden Summanden größer als Null sein.)

Die Zahl in Zelle **X4** wird mit dem Faktor **FactorDirC2** multipliziert und in der Formel für die Rangzahl in Zelle **V4** berücksichtigt.

7.7 Direkter Vergleich aller vier Mannschaften

Sind alle vier Mannschaften punktgleich, so werden die vier Mannschaften zunächst nach den Kriterien eines Direktvergleichs (Punktzahl, Tordifferenz und Anzahl der erzielten Tore) in eine Reihenfolge gebracht. Dabei können verschiedene Spezialfälle auftreten:

1. Es können ein oder zwei weitere Direktvergleiche jeweils zweier Mannschaften erforderlich werden.
2. Es kann ein weiterer Direktvergleich dreier Mannschaften erforderlich werden.
3. Es kann ein weiterer Direktvergleich dreier Mannschaften erforderlich werden und innerhalb dieses Dreiervergleichs ein weiterer Direktvergleich zweier Mannschaften.

Wir betrachten zunächst **Punkt 1**:

Weiterer Direktvergleich zweier Mannschaften

In Zelle **AF16** tragen wir folgende Formel ein:

```
=UND($Q$4=$Q$5;$Q$4=$Q$6;$Q$4=$Q$7)
```

Dies ist eine einfache Prüfung, ob alle vier Mannschaften punktgleich sind. Als Ergebnis enthält die Zelle den Wert TRUE oder den Wert FALSE.

In Zelle **M18** tragen wir folgende Formel ein:

```
=WENN($AF$16;WENN(UND($AD$4=$AD$5;$AD$4<>$AD$6;$AD$4<>$AD$7);WENN(I18>H19;3;WENN(I18=H19;1;0));WENN(UND($AD$4<>$AD$5;$AD$4=$AD$6;$AD$4<>$AD$7);WENN(J18>H20;3;WENN(J18=H20;1;0));WENN(UND($AD$4<>$AD$5;$AD$4<>$AD$6;$AD$4=$AD$7);WENN(K18>H21;3;WENN(K18=H21;1;0));""))
```

Zur Erinnerung:

In den Zellen **AD4:AD7** stehen Wertungszahlen, die nur die Punktzahl, die Tordifferenz und die Anzahl der erzielten Tore beinhaltet (siehe Kap. 7.2).

Die Formel in Zelle **M18** beginnt mit **WENN(\$AF\$16;...)**. Das bedeutet: Wenn alle vier Mannschaften punktgleich sind, dann ...

Der Ausdruck **WENN(UND(\$AD\$4=\$AD\$5;\$AD\$4<>\$AD\$6;\$AD\$4<>\$AD\$7);...)** prüft, ob Mannschaft 1 mit Mannschaft 2 in Punktzahl, Tordifferenz und erzielten Toren übereinstimmt, aber mit keiner weiteren Mannschaft.

Wenn ja, ist ein direkter Vergleich von Mannschaft 1 und Mannschaft 2 fällig:

```
WENN(I18>H19;3;WENN(I18=H19;1;0))
```

Es werden 3 Punkte, 1 Punkt oder 0 Punkte in die Zelle **M18** eingetragen.

Wenn nein, wird geprüft, ob ein Direktvergleich von Mannschaft 1 mit Mannschaft 3 oder Mannschaft 4 notwendig ist. Gegebenenfalls wird auch in diesen Fällen die entsprechende Punktzahl des Direktvergleichs in die Zelle **M18** eingetragen.

Entsprechend lauten die Formeln in den Zellen darunter:

Zelle **M19**:

```
=WENN($AF$16;WENN(UND($AD$5=$AD$4;$AD$5<>$AD$6;$AD$5<>$AD$7);WENN(H19>I18;3;WENN(H19=I18;1;0));WENN(UND($AD$5<>$AD$4;$AD$5=$AD$6;$AD$5<>$AD$7);WENN(J19>I20;3;WENN(J19=I20;1;0));WENN(UND($AD$5<>$AD$4;$AD$5<>$AD$6;$AD$5=$AD$7);WENN(K19>I21;3;WENN(K19=I21;1;0));""));""
```

Zelle **M20**:

```
=WENN($AF$16;WENN(UND($AD$6=$AD$4;$AD$6<>$AD$5;$AD$6<>$AD$7);WENN(H20>J18;3;WENN(H20=J18;1;0));WENN(UND($AD$6<>$AD$4;$AD$6=$AD$5;$AD$6<>$AD$7);WENN(I20>J19;3;WENN(I20=J19;1;0));WENN(UND($AD$6<>$AD$4;$AD$6<>$AD$5;$AD$6=$AD$7);WENN(K20>J21;3;WENN(K20=J21;1;0));""));""
```

Zelle **M21**:

```
=WENN($AF$16;WENN(UND($AD$7=$AD$4;$AD$7<>$AD$5;$AD$7<>$AD$6);WENN(H21>K18;3;WENN(H21=K18;1;0));WENN(UND($AD$7<>$AD$4;$AD$7=$AD$5;$AD$7<>$AD$6);WENN(I21>K19;3;WENN(I21=K19;1;0));WENN(UND($AD$7<>$AD$4;$AD$7<>$AD$5;$AD$7=$AD$6);WENN(J21>K20;3;WENN(J21=K20;1;0));""));""
```

In Zelle **AA4** tragen wir die Formel **=WENN(\$M18="";0;\$M18)** ein und kopieren sie nach unten bis zur Zelle **AA7**. Die Punktzahlen aus dem Bereich **M18:M21** werden in die Zellen **AA4:AA7** übertragen, um sie dann in den endgültigen Bewertungszahlen im Bereich **V4:V7** zu berücksichtigen.

Wir betrachten nun **Punkt 2**:

Weiterer Direktvergleich dreier Mannschaften

In Zelle **AD18** tragen wir folgende Formel ein:

=WENN(UND(\$AF\$16;\$AD\$4=\$AD\$5;\$AD\$4=\$AD\$6;\$AD\$4<>\$AD\$7);W18;0)

Wenn alle vier Mannschaften punktgleich sind (**\$AF\$16**) und die ersten drei die gleiche Tordifferenz und die gleiche Anzahl erzielter Tore haben (**\$AD\$4=\$AD\$5;\$AD\$4=\$AD\$6;\$AD\$4<>\$AD\$7**), wird die schon in Kap. 7.5 besprochene Rangzahl aus dem Direktvergleich der drei betreffenden Mannschaften in die Zelle **AD18** kopiert. Ansonsten wird eine Null eingetragen.

Die Formel wird nach unten kopiert bis zur Zelle **AD20**.

Im Bereich **AD23:AD25** stehen nun die Formeln für den Fall, dass alle vier Mannschaften punktgleich sind und für die Mannschaften 1, 2 und 4 ein direkter Vergleich durchgeführt wird.

In Zelle **AD23** tragen wir die Formel

=WENN(UND(\$AF\$16;\$AD\$4=\$AD\$5;\$AD\$4=\$AD\$7;\$AD\$4<>\$AD\$6);W23;0)

ein und kopieren sie nach unten bis zur Zelle **AD25**.

Entsprechend sehen die restlichen Formeln aus:

In Zelle **AD28** tragen wir die Formel

=WENN(UND(\$AF\$16;\$AD\$4=\$AD\$6;\$AD\$4=\$AD\$7;\$AD\$4<>\$AD\$5);W28;0)

ein und kopieren sie nach unten bis zur Zelle **AD30**.

In Zelle **AD33** tragen wir die Formel

=WENN(UND(\$AF\$16;\$AD\$5=\$AD\$6;\$AD\$5=\$AD\$7;\$AD\$5<>\$AD\$4);W33;0)

ein und kopieren sie nach unten bis zur Zelle **AD35**.

Um diese Rangzahlen im Bereich **AD18:AD35** bei den endgültigen Bewertungszahlen (im Bereich **V4:V7**) zu berücksichtigen, tragen wir in Zelle **Y4** die Formel **=SUMMEWENNS(\$AD\$18:\$AD\$35;\$P\$18:\$P\$35;\$P4)** ein und kopieren sie nach unten bis zur Zelle **Y7**.

Die Formel summiert für die erste Mannschaft (Mannschaftsname in Zelle **P4**) alle Zahlen des Bereichs **AD18:AD35**. Für diese Mannschaft kann höchstens einmal ein Wert größer als Null vorkommen.

Wir betrachten nun **Punkt 3**:

Weiterer Direktvergleich dreier Mannschaften plus weiterer Direktvergleich zweier Mannschaften

Einen solchen Fall kann es tatsächlich geben – siehe „Anhang: Spezialfall“. Allerdings müsste nach meinen Überlegungen dieser zusätzliche Zweiervergleich immer unentschieden ausgehen, so dass dieser Punkt 3 eigentlich nicht behandelt werden müsste. Aber man wird oft von Dingen überrascht, die man nicht für möglich gehalten hat. Da der Zusatzaufwand sehr gering ist, bauen wir deshalb zur Sicherheit auch die Formeln für diesen Fall ein:

In Zelle **AE18** tragen wir folgende Formel ein:

=WENN(AD18>0;Y18;"")

AD18>0 bedeutet, dass der Fall „Vier punktgleiche Mannschaften und Direktvergleich dreier Mannschaften“

eingetreten ist. In diesem Fall wird das bereits errechnete Ergebnis des Zweiervergleichs aus der Zelle **Y18** kopiert, ansonsten bleibt die Zelle leer.

Die Formel wird nach unten bis zur Zelle **AD35** kopiert.

In Zelle **Z4** tragen wir die Formel **=SUMMEWENNS(\$AE\$18:\$AE\$35;\$P\$18:\$P\$35;\$P4)** ein und kopieren sie nach unten bis zur Zelle **Z7**.

Um nun die Ergebnisse in den drei Spalten des Bereichs **Y4:AA7** in die endgültigen Bewertungszahlen (Zellen **V4:V7**) einfließen zu lassen, dürfen wir nicht die bisherigen Wichtungsfaktoren für die Direktvergleiche von drei bzw. zwei Mannschaften (FactorDirC3 bzw. FactorDirC2) verwenden. Denn bei Punktgleichheit aller vier Mannschaften wird zuerst nach Tordifferenz und Anzahl erzielter Tore sortiert – das ist sozusagen der erste Direktvergleich – und dann erst nach den weiteren Direktvergleichen.

Wir müssen also neue Wichtungsfaktoren erschaffen, die kleiner als die Wichtungsfaktoren der Tordifferenz und der Toranzahl sind. Auf der anderen Seite müssen sie größer sein als der Wichtungsfaktor für Fair-Play.

Wir wählen:

FactorDirC43 = 0,01

FactorDirC42 = 0,001

Der Wichtungsfaktor für das Elfmeterschießen konkurriert damit nicht, weil es bei Punktgleichheit aller vier Mannschaften kein Elfmeterschießen gibt.

7.8 Rangliste - Teil 3

Vom Tabellenblatt **GrpA** erstellen wir jetzt fünf Kopien, die wir **GrpB**, **GrpC**, . . , **GrpF** nennen.

Eine Kopie eines Tabellenblattes kann man erstellen, wenn man mit der rechten Maustaste unten auf den Namen des Tabellenblattes klickt, dann aus dem Kontextmenü "Verschieben oder kopieren." wählt. Nicht das Häkchen bei "Kopie erstellen" vergessen!

Auf jeder Kopie müssen wir nur **eine einzige Zelle** ändern: die Zelle **C1**.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T
1		Group A																		
3		Match		Score	Points	Penalties shoot	valid										Points	GD	Forced	Accepted
4		Türkei	- Italien	0 - 3	0 3		1								1.	Türkei	0	-7	1	8
5		Wales	- Schweiz	1 - 1	1 1		1								2.	Italien	9	5	5	0
6		Türkei	- Wales	0 - 2	0 3		1								3.	Wales	4	1	3	2
7		Italien	- Schweiz	1 - 0	3 0		1								4.	Schweiz	4	1	4	3
8		Schweiz	- Türkei	3 - 1	3 0	:	1													
9		Italien	- Wales	1 - 0	3 0	:	1													

Für die Gruppe B tragen wir hier ein "B" ein, für die Gruppe C ein "C" usw.

Unsere Formeln in den Bereichen **B4:G9**, **J8:L9** und **P4:P7** bewirken, dass allein durch Änderung der Zelle **C1** die ganze Tabelle mit den Daten der richtigen Gruppe gefüllt wird.

7.9 Gruppentabelle auf dem Spielplan

Auf dem Tabellenblatt „EURO“ tragen wir in die Zellen **C2**, **F2**, **I2**, **L2**, **O2** und **R2** die Formel **=Language!\$E\$100** ein. Die Leerzeichen in dem Text „ Pkt. Tore“ sorgen für die richtige Position der Überschriften „Pkt.“ und „Tore“.

Zelle **B3**: **=GrpA!P11**

Diese Formel wird nach unten kopiert bis zur Zelle **B6**.

Zelle **E3**: **=GrpB!P11**

Diese Formel wird nach unten kopiert bis zur Zelle **E6**.

Entsprechendes tragen wir für die restlichen sechs Gruppen ein.

Um die Punkte und die Tore anzuzeigen, tragen wir in die Zelle **C3** die Formel

= " &GrpA!\$Q11&" "&GrpA!\$S11&Language!\$E\$143&GrpA!\$T11

ein. Auch hier sorgen die eingefügten Leerzeichen für die richtige Positionierung. In **Language!\$E\$143** steht ein Trennzeichen für das Torverhältnis (Doppelpunkt oder Minus) umgeben von je einem Leerzeichen.

Auch diese Formel wird nach unten kopiert bis zur Zelle **C6**.

In die Zelle **F3** tragen wir die Formel

= " &GrpB!\$Q11&" "&GrpB!\$S11&Language!\$E\$143&GrpB!\$T11

ein. Entsprechendes tragen wir für die restlichen sechs Gruppen ein.

In der **Schlusstabelle** in den Zeilen 45 bis 49 sollen die Felder leer bleiben, wenn noch nicht alle Spiele in der betreffenden Gruppe eingetragen wurden.

Wir tragen in die Zellen **B41** und **B42** folgende Formeln ein:

Zelle **B45**: **=Language!\$E\$106**

Zelle **B46**: **=WENN(GrpA!\$M\$10=6;GrpA!\$P11;"")**

Die Formel in **B46** wird nach unten kopiert bis zur Zelle **B49**.

Die Schlusstabellen der anderen Gruppen haben die gleichen Formeln, nur statt „GrpA“ wird „GrpB“ usw. eingetragen.

Der aktuelle Stand des Spielplans ist in der Datei „**EURO_Tut_7.xlsx**“ zu finden.

8 KO-Phase

8.1 Bestimmung der vier besten Gruppendritten

Wir erzeugen ein neues Tabellenblatt mit dem Namen „**ThirdPlaced**“. Es wird folgendermaßen aussehen:

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
1			Gruppendritte														
2																	
3				Pts	GD	GF	GA	Wins	Coefficient	Grp		Fair play	Ranking				
4		1.	Wales	4	1	3	2	1	40.001.031,000081	A		0	81				
5		2.	Finnland	3	-2	1	3	1	29.998.011,000080	B		0	80				
6		3.	Ukraine	3	-1	4	5	1	29.999.041,000094	C		0	94				
7		4.	Tschechien	4	1	3	2	1	40.001.031,000082	D		0	82				
8		5.	Slowakei	3	-5	2	7	1	29.995.021,000078	E		0	78				
9		6.	Portugal	4	1	7	6	1	40.001.071,000087	F		0	87				
10				Pts	GD	GF	GA	Wins	Coefficient	Grp							
11		1.	Portugal	4	1	7	6	1	40.001.071,000087	F		3	A	4	7 : 6		
12		2.	Tschechien	4	1	3	2	1	40.001.031,000082	D		2	C	4	3 : 2		
13		3.	Wales	4	1	3	2	1	40.001.031,000081	A		0	D	4	3 : 2		
14		4.	Ukraine	3	-1	4	5	1	29.999.041,000094	C		1	F	3	4 : 5		
15		5.	Finnland	3	-2	1	3	1	29.998.011,000080	B				3	1 : 3		
16		6.	Slowakei	3	-5	2	7	1	29.995.021,000078	E				3	2 : 7		
17																	
18									Group combination:	A C D F							
19																	
20																	
21									Red point:	0							
22																	

Im Bereich **C4:C9** sind alle Gruppendritten aufgelistet. In Zelle **C4** steht z.B. die Formel **=GrpA!P\$13**.

Im Bereich **D4:H9** werden für jede Mannschaft die erreichten Punkte, die Tordifferenz, die erzielten Tore, die Gegentore und die Anzahl der Siege aus den Tabellenblättern für die jeweilige Gruppe kopiert. Die Formel in Zelle **D4** lautet z.B. **=GrpA!Q\$13**.

In Zelle **L4** tragen wir die Formel

**=SUMMEWENNS(FairPlayPoints1;FairPlayTeams1;\$C4)
+SUMMEWENNS(FairPlayPoints2;FairPlayTeams2;\$C4)**

ein und kopieren sie nach unten bis zur Zelle **L9**.

Diese Formeln ermitteln die Fair-Play-Wertungen für die einzelnen Mannschaften.

In Zelle **M4** tragen wir die Formel

=100-INDEX(Language!\$D\$5:\$D\$59;VERGLEICH(\$C4;Language!\$E\$5:\$E\$59;0);1)

ein und kopieren sie nach unten bis zur Zelle **M9**.

Hier soll eine Zahl stehen, die umso größer ist, je besser der in der Qualifikationsrunde erreichte Rang ist (European Qualifiers Ranking). Hier wird vorausgesetzt, dass die Rangzahlen auf dem Tabellenblatt ‚Language‘ in der Spalte D korrekt sind. Sie müssen nicht sortiert sein.

Der Ausdruck **VERGLEICH(\$C4;Language!\$E\$5:\$E\$59;0)** gibt die Position des Mannschaftsnamens (**\$C4**) auf dem Tabellenblatt ‚Language‘ zurück – zum Beispiel 19 für Wales. Die INDEX-Funktion schaut dann nach, welche Rangzahl links neben dem Mannschaftsnamen steht. Das ist in diesem Fall wieder die 19, weil die Liste

hier sortiert ist. Damit eine kleinere Rangzahl zu einer größeren Bewertungszahl führt, wird die Rangzahl von 100 subtrahiert.

In Zelle **I4** tragen wir die Formel

=D4*FactorPts+E4*FactorGD+F4*FactorFor+H4*FactorWins+L4*FactorFairPlay+M4*FactorRank

ein und kopieren sie nach unten bis zur Zelle **I9**.

Hier werden Bewertungszahlen berechnet, die Punktzahl, Tordifferenz, erzielte Tore, Anzahl der Siege, Fair-Play und Ranking berücksichtigen. Aufgrund dieser Bewertungszahlen wird nun im Bereich **B11:J16** eine sortierte Liste der sechs Gruppendritten erzeugt. Die Sortierung erfolgt nach demselben Prinzip wie in Kap. 7.4 beschrieben.

In der Spalte J werden die Gruppen-Buchstaben mitsortiert, so dass bekannt ist, aus welchen Gruppen die vier besten Mannschaften kommen. In unserem Fall sind das die Gruppen F, D, A, C.

Um diese vier alphabetisch zu sortieren, tragen wir in Zelle **L11** die Formel

=SUMME(WENN(\$J\$11:\$J\$14 < \$J11;1))

ein und schließen die Eingabe mit Strg-Umschalt-Eingabe (Ctrl-Shift-Enter) ab. Damit wird diese Formel zu einer sog. Matrix-Formel und erscheint in der Eingabezeile in geschweiften Klammern. Wir kopieren die Formel nach unten bis zur Zelle **L14**.

Diese Formel zählt für jeden Buchstaben, wie viele kleinere in der Liste vorhanden sind. Das ergibt z.B. für den ersten Buchstaben F die Anzahl 3, weil drei der Buchstaben kleiner sind als F. Die Zahlen im Bereich **L11:L14** sagen also, in welcher Reihenfolge die Buchstaben anzuordnen sind.

In die Zelle **M11** tragen wir die Formel

=INDEX(\$J\$11:\$J\$14;VERGLEICH(ZEILE()-11;\$L\$11:\$L\$14;0))

ein und kopieren sie nach unten bis zur Zelle **M14**.

Diese Formel sortiert die vier Buchstaben in aufsteigender Reihenfolge. Der Ausdruck **ZEILE()-11** erzeugt die vier Zahlen 0, 1, 2 und 3. Der Ausdruck **VERGLEICH(ZEILE()-11;\$L\$11:\$L\$14;0)** gibt die Position dieser vier Zahlen in der Liste **L11:L14** zurück. Die INDEX-Funktion sucht dann aus der Liste **J11:J14** den entsprechenden Buchstaben heraus.

Die sortierten Buchstaben werden in Zelle **J18** mit Hilfe der Formel **=M11&" "&M12&" "&M13&" "&M14** zu einer Zeichenkette zusammengefasst.

Die Gruppentabelle im Bereich **N11:Q16** ist eigentlich überflüssig. Sie ist nur optischer Komfort.

Die Formel in Zelle **J21** wird im am Ende des [→Kapitels 9.1](#) besprochen. Sie lautet:

=WENN(UND(KÜRZEN(\$I\$14;4)=KÜRZEN(\$I\$15;4);Distinctness!\$G\$13>0);1;0)

Wir erzeugen ein neues Tabellenblatt mit dem Namen „**AssignThird**“ (Zuordnung der Dritten).

Auf diesem Tabellenblatt gibt es keine Formel. Wir übertragen einfach die von der UEFA vorgegebene Tabelle für die Zuordnung der Gruppendritten im Achtelfinale. Dabei ist darauf zu achten, dass in der Spalte C zwischen jedem Buchstaben genau ein Leerzeichen steht. Denn dann können wir mit Hilfe der bedingten Formatierung die aktuelle Gruppenkombination farbig hervorheben.

Als Formel für die bedingte Formatierung verwenden wir in Zeile 8:

=C8=ThirdPlaced!\$J\$18

Diese Formatierung kann mit „Format übertragen“ auf die anderen Zeilen übertragen werden.

Das Tabellenblatt ‚**Distinctness**‘ (Unterscheidbarkeit, Klarheit) wird erst im Kapitel 9.2 besprochen. In diesem Kapitel benötigen wir allerdings schon den Bezug auf die Zelle **G13**. Deshalb erzeugen wir nun ein Tabellenblatt mit dem Namen ‚**Distinctness**‘ und tragen in die Zelle **G13** die folgende Formel ein:

=GrpA!\$M\$10+GrpB!\$M\$10+GrpC!\$M\$10+GrpD!\$M\$10+GrpE!\$M\$10+GrpF!\$M\$10

Die Formel gibt die Gesamtzahl aller vollständig eingetragenen Spielergebnisse der Gruppenphase zurück.

Auf dem Tabellenblatt ‚**EURO**‘ soll die Konstellation der Gruppendritten im Bereich **T45:Y49** angezeigt werden. Die Zellen **T45** und **U45** werden miteinander verbunden. Hier tragen wir die Formel **=Language!\$E\$126** ein.

In Zelle **T46** tragen wir die Formel **=WENN(Distinctness!\$G\$13>0;ThirdPlaced!\$C11;"")** ein und kopieren sie nach unten bis zur Zelle **T49**. Hier erscheinen die vier qualifizierten Gruppendritten – jedoch erst, wenn mindestens ein Gruppenspiel stattgefunden hat (**Distinctness!\$G\$13>0**).

Die beiden nicht qualifizierten Mannschaften werden in den Zellen **W48** und **W49** angezeigt.

Zelle **W48**: **=WENN(Distinctness!\$G\$13>0;ThirdPlaced!\$C15;"")**

Zelle **W49**: **=WENN(Distinctness!\$G\$13>0;ThirdPlaced!\$C16;"")**

Zum Anzeigen der Punkte und der Tore tragen wir in Zelle **U46** die Formel

=WENN(Distinctness!\$G\$13>0;" "&ThirdPlaced!\$D11&" "&ThirdPlaced!\$F11&Language!\$E\$143 &ThirdPlaced!\$G11;"")

ein und kopieren sie nach unten bis zur Zelle **U49**.

Zelle **X48**:

=WENN(Distinctness!\$G\$13>0;" "&ThirdPlaced!\$D15&" "&ThirdPlaced!\$F15&Language!\$E\$143 &ThirdPlaced!\$G15;"")

Zelle **X49**:

=WENN(Distinctness!\$G\$13>0;" "&ThirdPlaced!\$D16&" "&ThirdPlaced!\$F16&Language!\$E\$143 &ThirdPlaced!\$G16;"")

Formeln für die Gruppenzugehörigkeit:

Zelle **V46**: **=WENN(Distinctness!\$G\$13>0;ThirdPlaced!\$J11;"")**

Die Formel wird nach unten kopiert bis zur Zelle **V49**.

Zelle **Y48**: **=WENN(Distinctness!\$G\$13>0;ThirdPlaced!\$J15;"")**

Die Formel wird nach unten kopiert in die Zelle **Y49**.

Die Zellen **W46** und **X46** werden miteinander verbunden. Mit der Formel

=WENN(Distinctness!\$G\$13>0;ThirdPlaced!\$J18;"")

wird hier die aktuelle Gruppenkombination angezeigt.

8.2 Achtelfinale

Für die Gruppenphase haben wir alle Match-Daten so weit vorbereitet, dass wir auf dem Tabellenblatt „EURO“ nur noch links oben neben jedes Spiel-Kästchen die laufende Nummer des Spiels eintragen brauchen. Die Spiel-Daten werden dann automatisch aus dem Tabellenblatt „Matches“ übernommen.

Dieses Prinzip werden wir auch im Achtelfinale anwenden.

Bei den anderen Spielen der KO-Phase (Viertelfinale, Halbfinale, Finale) wird es eine Ausnahme geben:

Die Mannschaftsnamen für die einzelnen Begegnungen werden nicht auf dem Tabellenblatt „Matches“ vorbereitet.

Eigentlich wäre es am schönsten, dieses Prinzip für alle Spiele bis zum Finale zu verwenden. Aber ab dem Viertelfinale hängen die Mannschaftsnamen von Eintragungen auf dem Tabellenblatt „EURO“ ab. Wir würden schwer lesbare Formeln mit unnötigem Rechenaufwand benötigen, um zuerst die Mannschaftsnamen auf dem Tabellenblatt „Matches“ zu erzeugen und dann auf das Tabellenblatt „EURO“ zu übertragen.

Auf dem Tabellenblatt „Matches“ sehen wir in den Zellen **C41** und **D41**, dass das Spiel Nr. 37 die Begegnung **2A** gegen **2B** ist. In den Spalten I und J sollen nun die entsprechenden Mannschaftsnamen stehen - aber nicht statisch, sondern in Abhängigkeit von den Einträgen in den Spalten C und D - so wie bei den Gruppenspielen.

Dazu legen wir im Bereich **M41:N56** eine kleine Hilfstabelle an. Sie hat folgende Einträge:

	M	N
41	1A	=EURO!\$B\$46
42	2A	=EURO!\$B\$47
43	1B	=EURO!\$E\$46
44	2B	=EURO!\$E\$47
45	1C	=EURO!\$H\$46
46	2C	=EURO!\$H\$47
47	1D	=EURO!\$K\$46
48	2D	=EURO!\$K\$47
49	1E	=EURO!\$N\$46
50	2E	=EURO!\$N\$47
51	1F	=EURO!\$Q\$46
52	2F	=EURO!\$Q\$47
53	3A/D/E/F	*1
54	3D/E/F	*2
55	3A/B/C/D	*3
56	3A/B/C	*4

Der Mannschaftsname für **1A** steht in der Zelle **EURO!\$B\$46**, der Mannschaftsname für **2A** in der Zelle **EURO!\$B\$47** usw.

Die letzten vier Formeln sind:

*1:

```
=WENN(Distinctness!$G$13>0;SVERWEIS(SVERWEIS(ThirdPlaced!$J$18;AssignThird!$C$8:$G$22;2;0);WAHL({1.2};ThirdPlaced!$J$4:$J$9;ThirdPlaced!$C$4:$C$9);2;0);"")
```

*2:

```
=WENN(Distinctness!$G$13>0;SVERWEIS(SVERWEIS(ThirdPlaced!$J$18;AssignThird!$C$8:$G$22;3;0);WAHL({1.2};ThirdPlaced!$J$4:$J$9;ThirdPlaced!$C$4:$C$9);2;0);"")
```

*3:

```
=WENN(Distinctness!$G$13>0;SVERWEIS(SVERWEIS(ThirdPlaced!$J$18;AssignThird!$C$8:$G$22;4;0);WAHL({1.2};ThirdPlaced!$J$4:$J$9;ThirdPlaced!$C$4:$C$9);2;0);"")
```

*4:

```
=WENN(Distinctness!$G$13>0;SVERWEIS(SVERWEIS(ThirdPlaced!$J$18;AssignThird!$C$8:$G$22;5;0);WAHL({1.2};ThirdPlaced!$J$4:$J$9;ThirdPlaced!$C$4:$C$9);2;0);"")
```

Diese vier Formeln unterscheiden sich nur in einer einzigen Zahl (rote Schrift).

Mit dem Ausdruck **WENN(Distinctness!\$G\$13>0;...)** wird abgefragt, ob mindestens ein Spielergebnis der Gruppenphase eingetragen wurde. Wenn nicht, soll hier für die Gruppendritten zunächst nichts eingetragen werden.

Andernfalls wird ein verschachtelter SVERWEIS durchgeführt:

```
SVERWEIS(SVERWEIS(ThirdPlaced!$J$18;AssignThird!$C$8:$G$22;2;0);WAHL({1.2};ThirdPlaced!$J$4:$J$9;ThirdPlaced!$C$4:$C$9);2;0)
```

Der innere Ausdruck **SVERWEIS(ThirdPlaced!\$J\$18;AssignThird!\$C\$8:\$G\$22;2;0)** liest die aktuelle Gruppenkombination (z. B. A C D F) aus der Zelle **ThirdPlaced!\$J\$18** und schaut auf dem Tabellenblatt ‚AssignThird‘ nach, welcher Buchstabe in der Spalte D steht – hier z. B. der Buchstabe F.

Der Ausdruck **WAHL({1.2};ThirdPlaced!\$J\$4:\$J\$9;ThirdPlaced!\$C\$4:\$C\$9)** erzeugt für die äußere SVERWEIS-Funktion eine Suchmatrix, die aus zwei Spalten zusammengesetzt wird: **ThirdPlaced!\$J\$4:\$J\$9** und **ThirdPlaced!\$C\$4:\$C\$9**. In der Spalte I wird nach dem Buchstaben F gesucht und in der Spalte C wird der Name ‚Portugal‘ gefunden.

Wenn die rote **2** in der Formel nun eine **3** wird, findet die innere SVERWEIS-Funktion auf dem Tabellenblatt ‚AssignThird‘ den Buchstaben D, und die äußere SVERWEIS-Funktion findet auf dem Tabellenblatt ‚ThirdPlaced‘ den Namen ‚Tschechien‘.

In Zelle **I41** tragen wir die Formel **=SVERWEIS(C41;\$M\$41:\$N\$56;2;0)** ein und kopieren sie nach unten und nach rechts über den ganzen Bereich **I41:J48**. Es erscheinen nun die Mannschaftsnamen der Achtelfinalbegegnungen.

Die Überschriften „**Achtelfinale 1**“, „**Achtelfinale 2**“, usw. sind nicht offiziell vorgegeben. Wir wählen sie nach der zeitlichen Reihenfolge der Achtelfinalsplele. Man kann die Überschriften aber auch genauso gut von links nach rechts durchnummerieren.

Um eine Anpassung der zeitlichen Reihenfolge an künftige Europameisterschaftsturniere maximal zu vereinfachen, tragen wir in die Zellen **K41** bis **K48** die Zahlen von 1 bis 8 in der gewünschten Reihenfolge ein.

Den Rest sollen Formeln erledigen.

Bei der Europameisterschaft 2020/2021 fand das Spiel Nr. **37** (Achtelfinalspiel **2A-2B**) zuerst statt, es bekommt die Nummer 1.

Das Spiel Nr. **38** (Achtelfinalspiel **1A-2C**) war das zweite, es bekommt die Nummer 2.

Das Spiel Nr. **39** (Achtelfinalspiel **1C-3D/E/F**) ist das dritte, es bekommt die Nummer 3.

Auf diese Weise ergibt sich die zeitliche Reihenfolge 1, 2, 3, 4, 5, 6, 7, 8.

Die Numerierung der Spiele (37 – 44) stimmt hier mit der zeitlichen Reihenfolge überein. Dies muss nicht immer so sein.

Nun füllen wir den Spielplan mit den Daten des Achtelfinales.

Auf dem Tabellenblatt „EURO“ tragen wir in Zelle **B51** die Formel

=SVERWEIS(SVERWEIS(A52;Matches!\$B\$41:\$K\$48;10;0);Language!\$D\$108:\$E\$115;2;0) ein.

Die innere SVERWEIS-Funktion **SVERWEIS(A52;Matches!\$B\$41:\$K\$48;10;0)** liest in Zelle **A52** die Spielnummer „41“ und schaut dann auf dem Tabellenblatt ‚Matches‘ nach, welche Zahl für dieses Spiel in der Spalte K eingetragen ist. Sie findet die Zahl „5“.

Die äußere SVERWEIS-Funktion schaut nun auf dem Tabellenblatt ‚Language‘ im Bereich **\$D\$108:\$E\$115** nach, welche Überschrift zur Zahl „5“ gehört, und findet dort die sprachabhängige Überschrift für das erste Achtelfinalspiel, in deutsch also zum Beispiel „Achtelfinale 5“. Die Idee, nur das Wort „Achtelfinale“ zu übersetzen und eine 5 anzuhängen, kann eventuell zu Konflikten mit Sprachen führen, bei denen die 5 nicht am Ende steht.

Die Formeln für den Bereich **B52:C54** kopieren wir aus einem der Gruppenspiele: also zum Beispiel den Bereich **B38:C40** markieren, kopieren mit „**Strg C**“, dann auf die Zelle **B52** klicken und mit „**Start -> Einfügen -> Formeln**“ einfügen.

In Zelle **B56** tragen wir ein: **=Language!\$E\$125** („Elfmeterschießen:“). Die Zellen **B56** und **C56** werden miteinander verbunden.


Sie bekommen eine bedingte Formatierung (Schriftfarbe rot) mit der Formel **=UND(B55=C55;B55<>"";C55<>"").**

In Zeile 58 tragen wir die folgenden beiden Formeln ein:

Zelle **B58**: **=SVERWEIS(A52;Matches!\$B\$4:\$K\$58;2;0)**

Zelle **C58**: **=SVERWEIS(A52;Matches!\$B\$4:\$K\$58;3;0)**

Nun markieren wir den Bereich **B51:C58** und kopieren die Formeln in die anderen Achtelfinalspiele.

Erläuterung zu der Reihenfolge der Achtelfinals Spiele auf dem Spielplan:

Wenn wir die Begegnungen vom Finale aus rückwärts verfolgen, sehen wir im Plan der UEFA, dass im Finale der Gewinner des Spiels 49 gegen den Gewinner des Spiels 50 spielt. Im Spiel 49 spielt der Gewinner des Spiels 45 gegen den Gewinner des Spiels 46. Im Spiel 45 spielt der Gewinner des Spiels 41 gegen den Gewinner des Spiels 42.

So erhalten wir für die Reihenfolge der Achtelfinals Spiele auf dem Spielplan: 41, 42, 40, 38, 39, 37, 43, 44.

8.3 Viertelfinale, Halbfinale und Finale

Für das Viertelfinale tragen wir in die Zellen **K50:K53** des Tabellenblattes ‚Matches‘ die Zahlen 1, 2, 3, 4 ein. Sie ergeben sich aus der zeitlichen Reihenfolge der Viertelfinals Spiele.

Da die Reihenfolge der Halbfinalspiele und des Finales immer gleich ist, tragen wir in diesem Fall die entsprechenden Formeln ein:

Zelle **K55**: **=Language!\$E\$120**

Zelle **K56**: **=Language!\$E\$121**

Zelle **K58**: **=Language!\$E\$123**

Auf dem Tabellenblatt ‚EURO‘ tragen wir für das erste Viertelfinalspiel die folgenden Formeln ein:

Zelle **C61**: **=SVERWEIS(SVERWEIS(B62;Matches!\$B\$50:\$K\$53;10;0);Language!\$D\$116:\$E\$119;2;0)**

Zelle **C62**: **=SVERWEIS(B62;Matches!\$B\$4:\$K\$58;7;0)**

Zelle **C63**: **=SVERWEIS(B62;Matches!\$B\$4:\$K\$58;5;0)**

Zelle **C66**: **=Language!\$E\$125**

Zelle **C68**: **=SVERWEIS(B62;Matches!\$B\$50:\$K\$58;2;0)**

Zelle **E68**: **=SVERWEIS(B62;Matches!\$B\$50:\$K\$58;3;0)**

Wir markieren den Bereich **C61:E68** und kopieren ihn in die Kästchen der restlichen Viertelfinals Spiele und **des ersten Halbfinalspiels**. Die Formel für die Überschrift ist beim Halbfinale nicht mehr richtig und es erscheint der Error "#NV" ("No Value"). Ab dem Halbfinale haben wir gleichbleibende Überschriften mit einer anderen Formel. In Zelle **F71** tragen wir deshalb ein:

Zelle **F71**: **=SVERWEIS(E72;Matches!\$B\$50:\$K\$58;10;0)**

Wir markieren den Bereich **F71:H78** und kopieren ihn in die Kästchen des zweiten Halbfinalspiels und des Finales.

Die Zellen **C66:E66** werden verbunden und bekommen eine bedingte Formatierung für rote Schriftfarbe mit der Formel **=UND(C65=E65;C65<>"";E65<>"")**, die auf alle KO-Spiele ab Viertelfinale erweitert wird.

An dieser Stelle haben wir die Möglichkeit einer weiteren Optimierung:

Auf dem Tabellenblatt ‚EURO‘ können die **Nummern der Spiele der KO-Runde** (Achtelfinale, Viertelfinale und Halbfinale) durch Formeln **automatisch** erzeugt werden. Das hat den Vorteil, dass nach dem Ausfüllen des Match-Plans auf dem Tabellenblatt ‚Matches‘ alle Spiele der KO-Runde automatisch richtig angeordnet werden.

Auf dem Tabellenblatt ‚EURO‘ hat die Zelle **C68** den Inhalt „W41“. Dies bedeutet, dass das erste Achtelfinal-Spiel (ganz links) das Spiel Nr. 41 sein muss. Wir ersetzen deshalb die Zahl 41 in der Zelle **A52** durch die Formel **=RECHTS(\$C\$68;2)*1**.

Von dem Text „W41“ werden die letzten beiden Zeichen („41“) abgeschnitten. Die Multiplikation mit 1 verwandelt den abgeschnittenen Text in die **Zahl 41**. Diese Umwandlung ist notwendig, weil die VERWEIS-Funktionen im Bereich **B51:C58** den Text „41“ auf dem Tabellenblatt ‚Matches‘ nicht finden würden. Sie würden den Error-Code **#NV** zurückgeben.

Entsprechend verfahren wir mit den anderen Spielnummern der KO-Runde. Zum Beispiel tragen wir in Zelle **D52** die Formel **=RECHTS(\$E\$68;2)*1** ein. In Zelle **H62** steht dann die Formel **=RECHTS(\$H\$78;2)*1** und in Zelle **E72** die Formel **=RECHTS(\$L\$88;2)*1**.

In der Zelle **K82** steht die Zahl 51, da das Finale immer das Spiel Nr. 51 ist.

Nun fehlen nur noch die **Mannschaftsnamen**.

In Zelle **C64** tragen wir die folgende Formel ein:

```
=WENN(UND(B55<>"";C55<>""));WENN(B55>C55;B54;WENN(B55<C55;C54;WENN(UND(B57<>"";C57<>""));WENN(B57>C57;B54;WENN(B57<C57;C54;""));""));""))
```

Hier soll der Gewinner des Spiels Nr. 41 stehen. Die Formel besteht aus sechs geschachtelten WENN-Funktionen. Die erste WENN-Funktion prüft, ob das Spielergebnis in den Zellen **B55** und **C55** vollständig eingetragen ist. Wenn nicht, wird kein Mannschaftsname eingetragen. Andernfalls: die nächsten beiden WENN-Funktionen prüfen, ob eine der beiden Mannschaften gewonnen hat. In diesem Fall wird der Sieger eingetragen. Andernfalls (unentschieden): jetzt wird geprüft, ob das Ergebnis des Elfmeterschießens vollständig ist und kein Unentschieden eingetragen wurde. Entsprechend wird der Sieger oder nichts für den Mannschaftsnamen eingetragen.

Hier die restlichen Formeln für die Mannschaftsnamen:

Zelle **E64**:

```
=WENN(UND(E55<>"";F55<>""));WENN(E55>F55;E54;WENN(E55<F55;F54;WENN(UND(E57<>"";F57<>""));WENN(E57>F57;E54;WENN(E57<F57;F54;""));""));""))
```

Zelle **I64**:

```
=WENN(UND(H55<>"";I55<>""));WENN(H55>I55;H54;WENN(H55<I55;I54;WENN(UND(H57<>"";I57<>""));WENN(H57>I57;H54;WENN(H57<I57;I54;""));""));""))
```

Zelle **K64**:

```
=WENN(UND(K55<>"";L55<>""));WENN(K55>L55;K54;WENN(K55<L55;L54;WENN(UND(K57<>"";L57<>""));WENN(K57>L57;K54;WENN(K57<L57;L54;""));""));""))
```

Zelle **O64**:

```
=WENN(UND(N55<>"";O55<>""));WENN(N55>O55;N54;WENN(N55<O55;O54;WENN(UND(N57<>"";O57<>""))
```

`=WENN(N57>O57;N54;WENN(N57<O57;O54;"");""));"")`

Zelle **Q64**:

`=WENN(UND(Q55<>"";R55<>"");WENN(Q55>R55;Q54;WENN(Q55<R55;R54;WENN(UND(Q57<>"";R57<>"");WENN(Q57>R57;Q54;WENN(Q57<R57;R54;"");""));""));"")`

Zelle **U64**:

`=WENN(UND(T55<>"";U55<>"");WENN(T55>U55;T54;WENN(T55<U55;U54;WENN(UND(T57<>"";U57<>"");WENN(T57>U57;T54;WENN(T57<U57;U54;"");""));""));"")`

Zelle **W64**:

`=WENN(UND(W55<>"";X55<>"");WENN(W55>X55;W54;WENN(W55<X55;X54;WENN(UND(W57<>"";X57<>"");WENN(W57>X57;W54;WENN(W57<X57;X54;"");""));""));"")`

Zelle **F74**:

`=WENN(UND(C65<>"";E65<>"");WENN(C65>E65;C64;WENN(C65<E65;E64;WENN(UND(C67<>"";E67<>"");WENN(C67>E67;C64;WENN(C67<E67;E64;"");""));""));"")`

Zelle **H74**:

`=WENN(UND(I65<>"";K65<>"");WENN(I65>K65;I64;WENN(I65<K65;K64;WENN(UND(I67<>"";K67<>"");WENN(I67>K67;I64;WENN(I67<K67;K64;"");""));""));"")`

Zelle **R74**:

`=WENN(UND(O65<>"";Q65<>"");WENN(O65>Q65;O64;WENN(O65<Q65;Q64;WENN(UND(O67<>"";Q67<>"");WENN(O67>Q67;O64;WENN(O67<Q67;Q64;"");""));""));"")`

Zelle **T74**:

`=WENN(UND(U65<>"";W65<>"");WENN(U65>W65;U64;WENN(U65<W65;W64;WENN(UND(U67<>"";W67<>"");WENN(U67>W67;U64;WENN(U67<W67;W64;"");""));""));"")`

Zelle **L84**:

`=WENN(UND(F75<>"";H75<>"");WENN(F75>H75;F74;WENN(F75<H75;H74;WENN(UND(F77<>"";H77<>"");WENN(F77>H77;F74;WENN(F77<H77;H74;"");""));""));"")`

Zelle **N84**:

`=WENN(UND(R75<>"";T75<>"");WENN(R75>T75;R74;WENN(R75<T75;T74;WENN(UND(R77<>"";T77<>"");WENN(R77>T77;R74;WENN(R77<T77;T74;"");""));""));"")`

8.4 Hinweis bei Eingabe eines unzulässigen Spielergebnisses

Ein Unentschieden ist in der KO-Phase nicht möglich, da das Spiel auf jeden Fall durch ein Elfmeterschießen entschieden wird. Wenn ein Unentschieden als Spielergebnis eingetragen wird, soll unter dem Spielergebnis ein Hinweis erscheinen.

Wir verbinden die Zellen **B59** und **C59** und tragen die folgende Formel ein:

`=WENN(UND(B55<>"";C55<>"";B55=C55;B57<>"";C57<>"";B57=C57);Language!E138;"")`

Diese Formel kopieren wir in die Zellen **E59**, **H59**, **K59**, **N59**, **Q59**, **T59** und **W59**.

Der Hinweis erscheint, wenn das Spielergebnis und das Ergebnis des Elfmeterschießens vollständig eingetragen wurden, aber kein Sieger daraus hervorgeht.

Wir verbinden die Zellen **C69**, **D69** und **E69** und tragen die folgende Formel ein:

`=WENN(UND(C65<>"";E65<>"";C65=E65;C67<>"";E67<>"";C67=E67);Language!E138;"")`

Diese Formel kopieren wir in die Zellen **I69**, **O69**, **U69**, **F79**, **R79** und **L89**.

Zum Schluss noch die Anzeige des neuen Weltmeisters:

Zelle **R82**:

=Language!\$E\$107

Zelle **R84**:

=WENN(UND(L85<>"";N85<>"");WENN(L85>N85;L84;WENN(L85<N85;N84;WENN(UND(L87<>"";N87<>"");WENN(L87>N87;L84;WENN(L87<N87;N84;""));"")));"")

Die Bereiche **R82:W83** und **R84:W85** werden jeweils zu einer Zelle verbunden.

Damit ist der Spielplan fertig. Der Rest (Kap. 9 und 10) ist zusätzlicher Komfort.

Der aktuelle Stand des Spielplans ist in der Datei „**EURO_Tut_8.xlsx**“ zu finden.

9 Optional: Die Sache mit dem roten Punkt

Sind zwei oder mehr Mannschaften am Ende der Gruppenphase aufgrund ihrer Punkte, Tore und den direkten Vergleichen nicht unterscheidbar, wird eine Entscheidung anhand der Fair-Play-Wertung oder durch die Rangliste der Qualifikationsrunde getroffen. Dann muss der Benutzer dieses Spielplans zusätzlich Bonuspunkte auf dem Tabellenblatt „FairPlay“ eintragen.

Damit das Eintreten eines solchen Falles deutlich erkennbar wird, werden die betroffenen Gruppen mit einem roten Punkt gekennzeichnet. Der rote Punkt verschwindet wieder, wenn die Reihenfolge durch das Eintragen von Bonuspunkten geklärt ist.

Um dies für den Benutzer noch transparenter zu machen, richten wir zusätzlich noch eine ins Auge fallende rote oder grüne Meldung ein, z. B.

„Gruppen mit Fair-Play-Wertung oder Losentscheid: A, C und F.“

oder

„Die Platzierung ist in allen Gruppen geklärt.“

9.1 Der rote Punkt

In der Zelle **X11** auf dem Tabellenblatt "GrpA" tragen wir folgende Formel ein:

```
=WENN(UND(ODER(KÜRZEN($V$11;4)=KÜRZEN($V$12;4);KÜRZEN($V$11;4)=KÜRZEN($V$13;4);KÜRZEN($V$11;4)=KÜRZEN($V$14;4);KÜRZEN($V$12;4)=KÜRZEN($V$13;4);KÜRZEN($V$12;4)=KÜRZEN($V$14;4);KÜRZEN($V$13;4)=KÜRZEN($V$14;4));$M$10>0);1;0)
```

Im Einzelnen:

KÜRZEN(\$V\$11;4) bedeutet, dass die Rangzahl in **V11** (z. B. 90055053,0000983) auf vier Nachkommastellen gekürzt wird. Damit werden der Rang in der Qualifikationsrunde und die Zeilennummer abgeschnitten und es bleibt die Rangzahl übrig, die alle UEFA-Kriterien von der Punktzahl bis zur Fair-Play-Wertung enthält.

Nennen wir diese gekürzten Rangzahlen mal R1, R2, R3 und R4, dann besagt die Formel:

Wenn mindestens zwei von diesen Rangzahlen gleich sind (R1=R2 oder R1=R3 oder ... R3=R4) und außerdem mindestens ein Spielergebnis in dieser Gruppe eingetragen ist (**\$M\$10>0**), dann erscheint hier in der Zelle **X11** eine 1, ansonsten eine 0.

Steht in der Zelle **X11** eine 1, soll in der betreffenden Gruppe ein roter Punkt angezeigt werden.

Die Formel in **GrpA!X11** wird nun in die Tabellenblätter „GrpB“, „GrpC“ usw. kopiert.

Auf dem Tabellenblatt „EURO“ verbinden wir die Zellen **B8** und **C8**. Hier soll der rote Punkt erscheinen.

Deshalb tragen wir hier die Formel **=WENN(GrpA!\$X\$11>0;"•";"")** ein.

Den Punkt kann man zuerst mit „Einfügen → Symbol“ in irgendeiner Zelle erzeugen und dann mit Kopieren und Einfügen in die Formel einfügen.

Auf dem Tabellenblatt „EURO“ wird die Formel in **B8** nach **E8**, **H8** usw. kopiert. „GrpA“ wird in „GrpB“, „GrpC“ usw. geändert.

In Zelle **T44** tragen wir die Formel **=WENN(ThirdPlaced!\$J\$21>0;"•";"")** ein. Bei den Gruppendritten spielt die Reihenfolge der ersten vier keine Rolle. Es kommt nur darauf an, dass die richtigen vier Mannschaften an den

ersten vier Positionen stehen. Um festzustellen, ob die Situation unklar ist oder nicht, reicht es deshalb zu prüfen, ob die Platzierungen auf dem vierten und fünften Platz eindeutig geklärt sind.

Die Formel in der Zelle **ThirdPlaced!\$J\$21** lautet deshalb:

=WENN(UND(KÜRZEN(\$I\$14;4)=KÜRZEN(\$I\$15;4);Distinctness!\$G\$13>0);1;0)

Stimmen die Bewertungszahlen in **\$I\$14** und **\$I\$15** in allen Stellen bis zur vierten Nachkommastelle überein, bedeutet dies, dass die Mannschaften auf Platz 4 und Platz 5 in Punkten, Tordifferenz, Anzahl erzielter Tore und Anzahl der Gewinne übereinstimmen und auch in der Fair-Play-Wertung keine unterschiedlichen Zahlen eingetragen wurden. In diesem Fall würde das letzte Kriterium, die Position im European Qualifiers Ranking, entscheiden und ein roter Punkt wird über der Liste der Gruppendritten angezeigt. Die zusätzliche Bedingung **Distinctness!\$G\$13>0** sorgt dafür, dass der rote Punkt nur dann erscheint, wenn mindestens ein Spielergebnis eingetragen wurde.

9.2 Die Meldung

Wir füllen nun das Tabellenblatt „Distinctness“ aus. Es soll so aussehen:

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Distinctness of the ranks within the groups												
2													
3		Group	Distinctness	Number	Not clear	Indicator	Position	Text					
4		A	0	3		0	0				Gruppen mit Fair-Play-Wertung: B, C und E.		
5		B	1		B	1	1	B,			Die Platzierung ist in allen Gruppen geklärt.		
6		C	1		C	1	2	C und					
7		D	0			0	2						
8		E	1		E	1	3	E					
9		F	0			0	3						
10													
11							Full text:	B, C und E					
12													
13					Number of all games:		36						
14													

In die Zellen **B4** bis **B9** schreiben wir die Buchstaben von A bis F.

In die Zellen **C4** bis **C9** tragen wir die Formeln **=GrpA!\$X\$11**, **=GrpB!\$X\$11**, ... , **=GrpF!\$X\$11** ein.

Damit haben wir eine Übersicht, in welchen Gruppen die Fair-Play-Wertung bzw. die Rangliste der Qualifikationsrunde entscheidet. Das sind alle Gruppen, bei denen in der Spalte C eine 1 steht - in der Abbildung oben also die Gruppen B, C und E.

In diesem Fall soll auf dem Spielplan eine Meldung erscheinen, die folgenden Text hat:

Gruppen mit Fair-Play-Wertung: B, C und E.

Der erste Teil der Meldung (**Gruppen mit Fair-Play-Wertung:**) ist ein fester Text. Der zweite Teil (**B, C und E.**) ist ein variabler Text, der je nach Situation anders aussieht. Er kann z. B. auch heißen: **B und E.**

Um diesen zweiten Teil des Textes zu erzeugen, tragen wir zuerst in die Zelle **D4** eine Formel für die Anzahl der Gruppen mit Fair-Play-Wertung/Losentscheid ein: **=ZÄHLENWENN(\$C\$4:\$C\$9;">0")** .

In der Spalte E tragen wir den Buchstaben der Gruppe ein, wenn die Gruppe nicht eindeutig ist.

Zelle **E4**: **=WENN(\$C4>0;\$B4;"")**

Die Formel wir nach unten kopiert bis **E9**.

In die Zellen **F4** und **G4** tragen wir folgende Formeln ein:

Zelle **F4**: **=LÄNGE(\$E4)**

Zelle **G4**: **=SUMME(\$F\$4:\$F4)**

und kopieren beide Formeln nach unten bis zur Zeile 9. Wichtig ist, dass im Ausdruck **\$F\$4:\$F4** vor der letzten 4 kein Dollarzeichen steht. Aus **\$F4** wird beim Kopieren **\$F5** usw.

Nebenbemerkung: Die Spalte F ist nur eine Hilfsspalte, um die Verwendung der Funktion SUMMENPRODUKT zu vermeiden, die sich vom Wesen her wie eine Matrix-Formel verhält und deshalb schwerer verständlich ist. Man könnte nämlich auch die Spalte F weglassen und in Spalte G die Formel =SUMMENPRODUKT(LÄNGE(\$E\$4:\$E4)) nach unten kopieren.

Die Zahlen in Spalte G sagen uns jetzt, an welcher Position jede der drei nicht eindeutigen Gruppen steht. Das B in Zeile 5 steht an erster Position. An der 3 in Zeile **G9** sehen wir, dass es noch zwei weitere Gruppen gibt. Deshalb muss hinter dem B ein Komma folgen. Das D in Zeile 7 steht an zweiter Position. An der 3 in Zeile **G9** sehen wir, dass dies die vorletzte Position ist und hinter dem D ein „und“ folgen muss. Das E in Zeile 8 steht an dritter Position. An der 3 in Zeile **G9** sehen wir, dass dies die letzte Position ist. Nach dem E muss nichts mehr folgen. Der abschließende Punkt wird in der Formel in Zelle **J4** gesetzt.

Diese Logik ist in den Formeln in Spalte H zusammengefasst.

Zelle **H4**: **=WENN(\$E4<>"";\$E4&WENN(\$G4=\$G\$9;"";WENN(\$G4<\$G\$9-1;" ";Language!\$E\$137));"")**

Die Formel wird nach unten kopiert. Der Ausdruck **Language!\$E\$137** ist das Wort "und" in der gewählten Sprache.

In Zelle **H11** setzen wir alle Teile zu einem Text zusammen:

Zelle **H11**: **=\$H\$4&\$H\$5&\$H\$6&\$H\$7&\$H\$8&\$H\$9**

In Zelle **J4** steht dann die gesamte Meldung: **=Language!\$E\$135&" "&\$H\$11&".**

Formel in der Zelle **J5**: **=Language!\$E\$136**

Die Meldung in **J5** erscheint, wenn in **D4** eine Null steht.

Auf dem Tabellenblatt „EURO“ ist die Zeile 7 für die Meldung vorgesehen.

Wir verbinden die Zellen **B7:R7** zu einer Zelle und zentrieren ihren Inhalt.

In Zelle **B7** tragen wir die folgende Formel ein:

=WENN(Distinctness!\$G\$13=0;"";WENN(Distinctness!\$D\$4>0;Distinctness!\$J\$4;Distinctness!\$J\$5))

Die erste Bedingung **Distinctness!\$G\$13=0** bewirkt, dass keine Meldung erscheint, wenn im Spielplan noch keine Ergebnisse eingetragen wurden.

Die Zelle **B7** bekommt eine grüne Schriftfarbe. Mit Hilfe der **bedingten Formatierung** ändert sich die Schriftfarbe in Rot, wenn in einer der Gruppen eine ungeklärte Rangfolge vorliegt. Als Formel für die Bedingung tragen wir ein:

=Distinctness!\$D\$4>0 .

Der aktuelle Stand des Spielplans ist in der Datei „**EURO_Tut_9.xlsx**“ zu finden.

10 Optional: Zusätzliche Übersicht über die direkten Vergleiche

Diese Übersicht soll insbesondere beim direkten Vergleich dreier Mannschaften die Vergleichstabelle der drei betroffenen Mannschaften zeigen, um transparent zu machen, wie in diesem Fall die Rangreihenfolge zustande kommt. Aber auch der direkte Vergleich zweier Mannschaften wird in dieser Übersicht angezeigt.

Wir erzeugen ein neues Tabellenblatt mit dem Namen „Direct comparisons“ (Direkte Vergleiche).

Es soll folgendermaßen aussehen:

Direkte Vergleiche															Clear?	Row numbers	Col num	Dir. Comp. (3)			
Gruppe A				Gesamt		Dir. Vergl.(2)		Dir. Vergl.(3)			Fair-Play/Los	Nicht klar	Points	GD	For	Against					
	Pkt	TD	Tore	Siege	Pkt	Efm	Pkt	TD	Tore												
1. Italien	9	5	5 - 0	3	1					0		0	0	0	0	0					
2. Schweiz	4	1	4 - 3	1	1					0		0	0	0	0	0					
3. Wales	4	1	3 - 2	1	1					0		0	0	0	0	0					
4. Türkei	0	-7	1 - 8	0						0		0	0	0	0	0					
Gesamt											0	0	0	0	0	0	0				
Gruppe B																					
	Pkt	TD	Tore	Siege	Pkt	Efm	Pkt	TD	Tore	Fair-Play/Los	Nicht klar	Points	GD	For	Against						
1. Belgien	9	6	7 - 1	3						0		3	2	4	2						
2. Dänemark	3	1	5 - 4	1			3	2	4 - 2	0		3	0	1	1						
3. Finnland	3	-2	1 - 3	1			3	0	1 - 1	0		3	-2	2	4						
4. Russland	3	-5	2 - 7	1			3	-2	2 - 4	0		0	3	0	0						
Gesamt											0	6	0	0	2						
Gruppe C																					
	Pkt	TD	Tore	Siege	Pkt	Efm	Pkt	TD	Tore	Fair-Play/Los	Nicht klar	Points	GD	For	Against						
1. Niederlande	9	6	8 - 2	3						0		0	0	0	0						
2. Österreich	6	1	4 - 3	2						0		0	0	0	0						
3. Ukraine	3	-1	4 - 5	1						0		0	0	0	0						
4. Nordmazedonien	0	-6	2 - 8	0						0		0	0	0	0						
Gesamt											0	0	0	0	0	0					
Gruppe D																					
	Pkt	TD	Tore	Siege	Pkt	Efm	Pkt	TD	Tore	Fair-Play/Los	Nicht klar	Points	GD	For	Against						
1. England	7	2	2 - 0	2						0		0	0	0	0						
2. Kroatien	4	1	4 - 3	1	1					0		0	0	0	0						
3. Tschechien	4	1	3 - 2	1	1					0		0	0	0	0						
4. Schottland	1	-4	1 - 5	0						0		0	0	0	0						
Gesamt											0	0	0	0	0	0					
Gruppe E																					
	Pkt	TD	Tore	Siege	Pkt	Efm	Pkt	TD	Tore	Fair-Play/Los	Nicht klar	Points	GD	For	Against						
1. Schweden	7	2	4 - 2	2						0		0	0	0	0						
2. Spanien	5	5	6 - 1	1						0		0	0	0	0						
3. Slowakei	3	-5	2 - 7	1						0		0	0	0	0						
4. Polen	1	-2	4 - 6	0						0		0	0	0	0						
Gesamt											0	0	0	0	0	0					
Gruppe F																					
	Pkt	TD	Tore	Siege	Pkt	Efm	Pkt	TD	Tore	Fair-Play/Los	Nicht klar	Points	GD	For	Against						
1. Frankreich	5	1	4 - 3	1						0		0	0	0	0						
2. Deutschland	4	1	6 - 5	1	3					0		0	0	0	0						
3. Portugal	4	1	7 - 6	1	0					0		0	0	0	0						
4. Ungarn	2	-3	3 - 6	0						0		0	0	0	0						
Gesamt											0	0	0	0	0	0					

Die Spalten A bis X sind der für den Benutzer sichtbare Bereich. Die Spalten Y bis AI werden ausgeblendet.

Wir werden die Formeln für die Gruppe A entwickeln und dann den kompletten Bereich **B3:AI9** sieben Mal kopieren. In jeder dieser Kopien brauchen wir dann nur noch die Formel für die Überschrift (Zellen **C11**, **C19** usw.) anzupassen und in den Formeln mit Suchen/Ersetzen die Zeichenfolge „GrpA“ durch „GrpB“ bzw. „GrpC“ usw. zu ersetzen. Deshalb muss hier das Setzen bzw. Weglassen der Dollarzeichen genau beachtet werden.

10.1 Sichtbarer Bereich - Teil 1

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	
1	Direkte Vergleiche																							
2																								
3		●	Gruppe A	Gesamt				Dir. Vergl.(2)		Dir. Vergl.(3)														
4				Pkt	TD	Tore		Siege	Pkt	Elfm	Pkt	TD	Tore				Fair-Play/Los	Nicht klar						
5		1.	Italien	9	5	5 - 0		3									0							
6		2.	Schweiz	4	1	4 - 3		1	1								0							
7		3.	Wales	4	1	3 - 2		1	1								0							
8		4.	Türkei	0	-7	1 - 8		0									0							
9																								
10																								
11		●	Gruppe B	Gesamt				Dir. Vergl.(2)		Dir. Vergl.(3)														
12				Pkt	TD	Tore		Siege	Pkt	Elfm	Pkt	TD	Tore				Fair-Play/Los	Nicht klar						
13		1.	Belgien	9	6	7 - 1		3									0							
14		2.	Dänemark	3	1	5 - 4		1			3	2	4 - 2				0							
15		3.	Finnland	3	-2	1 - 3		1			3	0	1 - 1				0							
16		4.	Russland	3	-5	2 - 7		1			3	-2	2 - 4				0							
17																								
18																								
19			Gruppe C	Gesamt				Dir. Vergl.(2)		Dir. Vergl.(3)														
20				Pkt	TD	Tore		Siege	Pkt	Elfm	Pkt	TD	Tore				Fair-Play/Los	Nicht klar						
21		1.	Niederlande	9	6	8 - 2		3									0							
22		2.	Österreich	6	1	4 - 3		2									0							
23		3.	Ukraine	3	-1	4 - 5		1									0							
24		4.	Nordmazedonien	0	-6	2 - 8		0									0							
25																								

Wir beginnen mit der Überschrift:

Zelle **B1**: =Language!\$E\$85

Die Zellen **B1** bis **W1** werden miteinander verbunden.

Für die Überschriften in Zeile 3 tragen wir folgende Formeln ein:

Zelle **C3**: =Language!\$E\$93&" A"

Zelle **D3**: =Language!\$E\$101

Zelle **J3**: =Language!\$E\$104

Zelle **P3**: =Language!\$E\$105

Für die Überschriften in Zeile 4 tragen wir folgende Formeln ein:

Zelle **D4**: =Language!\$E\$95

Zelle **E4**: =Language!\$E\$96

Zelle **F4**: =Language!\$E\$97

Zelle **I4**: =Language!\$E\$98

Zelle **L4**: =Language!\$E\$95

Zelle **M4**: =Language!\$E\$99

Zelle **P4**: =Language!\$E\$95

Zelle **Q4**: =Language!\$E\$96

Zelle **R4**: =Language!\$E\$97

Zelle **V4**: =Language!\$E\$102

Zelle **W4**: =Language!\$E\$103

In die Zellen **B5** bis **B8** tragen wir die laufende Nummer ein. Die Mannschaftsnamen kopieren wir in sortierter Reihenfolge aus dem Tabellenblatt „GrpA“:

Zelle **C5**: =GrpA!\$P\$11
 Zelle **C6**: =GrpA!\$P\$12
 Zelle **C7**: =GrpA!\$P\$13
 Zelle **C8**: =GrpA!\$P\$14

Hier ist es z. B. für das spätere Kopieren der kompletten Gruppe wichtig, dass auch vor den Zahlen 11, 12, 13 und 14 ein Dollarzeichen steht.

Im Bereich **D5:I8** soll zunächst eine Kopie der Gruppentabelle zu sehen sein:

Zelle **D5**: =GrpA!\$Q\$11
 Zelle **D6**: =GrpA!\$Q\$12
 Zelle **D7**: =GrpA!\$Q\$13
 Zelle **D8**: =GrpA!\$Q\$14
 Zelle **E5**: =GrpA!\$R\$11
 Zelle **E6**: =GrpA!\$R\$12
 Zelle **E7**: =GrpA!\$R\$13
 Zelle **E8**: =GrpA!\$R\$14
 Zelle **F5**: =GrpA!\$S\$11
 Zelle **F6**: =GrpA!\$S\$12
 Zelle **F7**: =GrpA!\$S\$13
 Zelle **F8**: =GrpA!\$S\$14
 Zelle **H5**: =GrpA!\$T\$11
 Zelle **H6**: =GrpA!\$T\$12
 Zelle **H7**: =GrpA!\$T\$13
 Zelle **H8**: =GrpA!\$T\$14

Im Bereich **L5:L8** sollen die Punkte eines direkten Vergleichs zweier Mannschaften angezeigt werden. Findet kein direkter Vergleich zweier Mannschaften statt, bleiben die Felder leer.

Um festzustellen, ob für die betreffende Mannschaft ein direkter Vergleich mit **einer** anderen Mannschaft vorliegt, fügen wir zunächst weitere Formeln in die Gruppentabellen ein.

Auf dem **Tabellenblatt ‚GrpA‘** tragen wir in die Zelle **Z11** die Formel **=SVERWEIS(\$P11;\$E\$18:\$M\$21;8;0)<>""** ein und kopieren sie nach unten bis zur Zelle **Z14**. Diese Formel schaut in den Zellen **L18:L21** nach, ob dort für die betreffende Mannschaft ein Direktvergleich eingetragen ist. Entsprechend wird ‚WAHR‘ oder ‚FALSCH‘ eingetragen.

In die Zelle **AA11** tragen wir die Formel **=SVERWEIS(\$P11;\$E\$18:\$M\$21;9;0)<>""**

ein und kopieren sie nach unten bis zur Zelle **AA14**. Die Formel schaut in den Zellen **M18:M21** nach, ob dort ein Direktvergleich für die betreffende Mannschaft im Fall von vier punktgleichen Mannschaften eingetragen ist.

In die Zelle **AB11** tragen wir die Formel **=SUMMENPRODUKT(1*(LÄNGE(\$Z\$18:\$Z\$35)>0)*(\$P\$18:\$P\$35=\$P11))>0**

ein und kopieren sie nach unten bis zur Zelle **AB14**. Die Formel prüft, ob ein Direktvergleich zweier Mannschaften für die betreffende Mannschaft im Fall von drei punktgleichen Mannschaften vorliegt.

Bei der Funktion SUMMENPRODUKT werden die angegebenen Bereiche **\$Z\$18:\$Z\$35** und **\$P\$18:\$P\$35** Schritt für Schritt durchlaufen. Es werden also die Ausdrücke

1*(LÄNGE(\$Z\$18)>0)*(\$P\$18=\$P11)

1*(LÄNGE(\$Z\$19)>0)*(\$P\$19=\$P11)

...

1*(LÄNGE(\$Z\$35)>0)*(\$P\$35=\$P11)

berechnet und die Summe der einzelnen Ergebnisse gebildet.

Wenn also beispielsweise in der Zelle **Z19** eine Zahl (3, 1 oder 0) steht, hat der Ausdruck **LÄNGE(\$Z\$19)>0** den Wert WAHR (oder 1).

Da der Ausdruck **\$P\$19=\$P11** ebenfalls WAHR ist (in beiden Zellen steht ‚Italien‘), ergibt das Produkt **1*WAHR*WAHR = 1**.

Das Endergebnis **SUMMENPRODUKT(. . .)>0** ist dann ebenfalls WAHR.

Die nächste Formel in Zelle **AC11** arbeitet nach demselben Prinzip:

=SUMMENPRODUKT(1*(LÄNGE(\$AE\$18:\$AE\$35)>0)*(\$P\$18:\$P\$35=\$P11))>0

Sie prüft, ob für die betreffende Mannschaft ein Direktvergleich zweier Mannschaften vorliegt für den seltenen Fall, dass alle vier Mannschaften punktgleich sind und innerhalb des Direktvergleichs dreier Mannschaften noch ein weiterer Direktvergleich zweier Mannschaften durchgeführt wird.

In Zelle **AD11** tragen wir die Formel

=WENN(\$Z11;SVERWEIS(\$P11;\$E\$18:\$M\$21;8;0);WENN(\$AA11;SVERWEIS(\$P11;\$E\$18:\$M\$21;9;0);WENN(\$AB11;SUMMEWENNS(\$Z\$18:\$Z\$35;\$P\$18:\$P\$35;\$P11);WENN(\$AC11;SUMMEWENNS(\$AE\$18:\$AE\$35;\$P\$18:\$P\$35;\$P11);""))))

ein und kopieren sie nach unten bis zur Zelle **AD14**.

Für einen Direktvergleich zweier Mannschaften gibt es vier mögliche Fälle. Die Formeln in den Zellen **Z11**, **AA11**, **AB11** und **AC11** testen, ob einer dieser Fälle vorliegt. Die Formel in Zelle **AD11** schaut nun, ob einer dieser vier Fälle eingetreten ist. Wenn ja, trägt sie das Ergebnis des Direktvergleichs in die Zelle **AD11** ein.

Sie beginnt mit **WENN(\$Z11;SVERWEIS(\$P11;\$E\$18:\$M\$21;8;0);...**

Wenn also ein unmittelbarer Direktvergleich zweier Mannschaften im Bereich **L18:L21** eingetragen ist, kopiert sie diesen Wert in die Zelle **AD11**.

Andernfalls wird mit dem Ausdruck **WENN(\$AA11;SVERWEIS(\$P11;\$E\$18:\$M\$21;9;0);...** der nächste Fall abgefragt usw.

Die Ergebnisse im Bereich **AD11:AD14** werden auf das Tabellenblatt ‚Direct comparisons‘ übertragen.

Auf dem Tabellenblatt ‚Direct comparisons‘ tragen wir in die Zelle **L5** die Formel **=GrpA!\$AD\$11** ein und kopieren sie nach unten bis zur Zelle **L8**.

In den Zellen **M5:M8** des Tabellenblatts ‚Direct comparisons‘ soll dargestellt werden, ob ein Elfmeterschießen im letzten Gruppenspiel die Reihenfolge der Mannschaften entschieden hat. Auch hierfür werden die Ergebnisse zunächst auf dem Tabellenblatt der Gruppe kalkuliert und dann in die Zellen **M5:M8** übertragen.

Auf dem Tabellenblatt ‚GrpA‘ tragen wir in Zelle **AE11** die Formel

```
=WENN(ODER(UND($Y$11>0;ODER($B$8=$P11;$D$8=$P11));UND($Y$12>0;ODER($B$9=$P11;$D$9=$P11)));  
SUMMEWENN($B$8:$B$9;$P11;$J$8:$J$9)+SUMMEWENN($D$8:$D$9;$P11;$L$8:$L$9);"")
```

ein und kopieren sie nach unten bis zur Zelle **AE14**.

Hier wird das Ergebnis des Elfmeterschießens eingetragen, wenn die betreffende Mannschaft beteiligt ist.

Die Formel hat folgende Struktur:

```
=WENN(ODER(Bedingung1;Bedingung2);Ergebnis;"")
```

Bedingung1: **UND(\$Y\$11>0;ODER(\$B\$8=\$P11;\$D\$8=\$P11))**

Bedingung2: **UND(\$Y\$12>0;ODER(\$B\$9=\$P11;\$D\$9=\$P11))**

Ergebnis: **SUMMEWENN(\$B\$8:\$B\$9;\$P11;\$J\$8:\$J\$9)+SUMMEWENN(\$D\$8:\$D\$9;\$P11;\$L\$8:\$L\$9)**

\$Y\$11>0 bedeutet: Im fünften Gruppenspiel findet ein Elfmeterschießen statt (Erklärung in Kap. 7.2).

ODER(\$B\$8=\$P11;\$D\$8=\$P11) bedeutet, dass die betreffende Mannschaft (**\$P11**) daran beteiligt ist.

Die SUMMEWENN-Funktion addiert die Zahlen in den Bereichen **J8:J9** und **L8:L9**, wenn die betreffende Mannschaft (**\$P11**) im Bereich **B8:B9** oder **D8:D9** vorkommt.

Auf dem Tabellenblatt ‚Direct comparisons‘ tragen wir in Zelle **M5** die Formel **=GrpA!\$AE\$11** ein und kopieren sie nach unten bis zur Zelle **M8**. Damit erscheinen die Ergebnisse eines Elfmeterschießens im letzten Gruppenspiel im Bereich **M5:M8**.

Im Bereich **P5:T8** soll die Gruppentabelle für den direkten Vergleich dreier Mannschaften angezeigt werden. Dazu müssen im ausgeblendeten Bereich zuerst weitere Formeln vorbereitet werden (siehe Kap. 10.2). Wir werden diese Formeln später einsetzen und besprechen (siehe Kap. 10.3).

Im Bereich **V5:V8** soll gezeigt werden, ob ein Bonus für Fair-Play eingetragen wurde.

Zelle **V5**:

```
=SUMMEWENNS(FairPlayPoints1;FairPlayTeams1;$C5)  
+SUMMEWENNS(FairPlayPoints2;FairPlayTeams2;$C5)
```

Die Formel wird nach unten kopiert bis zur Zelle **V8**.

Im Bereich **W5:W8** soll ein roter Punkt erscheinen, wenn die Positionierung für die betreffende Mannschaft noch nicht klar ist. Die Formeln dafür werden wir später eintragen und besprechen (siehe Kap. 10.3).

10.2 Ausgeblendeter Bereich

	Y	Z	AA	AB	AC	AD	AE	AF	AG	AH	AI
1											
2											
3	Clear?		Row numbers				Col num	Dir. Comp. (3)			
4								Points	GD	For	Against
5	0		0	0	0	0		0	0	0	0
6	0		0	0	0	0		0	0	0	0
7	0		0	0	0	0		0	0	0	0
8	0		0	0	0	0					
9			0	0	0	0	0				
10											
11											
12											
13	0		0	0	0	0		3	2	4	2
14	0		0	1	0	0		3	0	1	1
15	0		0	2	0	0		3	-2	2	4
16	0		0	3	0	0					
17			0	6	0	0	2				
18											
19											
20											
21	0		0	0	0	0		0	0	0	0
22	0		0	0	0	0		0	0	0	0
23	0		0	0	0	0		0	0	0	0
24	0		0	0	0	0					
25			0	0	0	0	0				
26											

Um die direkten Vergleiche dreier Mannschaften darzustellen, benötigen wir eine Vorbereitung in dem ausgeblendeten Bereich (Spalten Y bis AI).

Es sind in jeder Gruppe vier verschiedene Dreiergruppen möglich. Die Tabellen für diese vier Dreiergruppen befinden sich auf dem Tabellenblatt ‚GrpA‘ im Bereich **O17:AD36**. Wenn kein direkter Vergleich dreier Mannschaften stattfindet, stehen in diesen vier Tabellen in den Spalten X und AD lauter Nullen. Nur im Fall eines direkten Vergleichs dreier Mannschaften treten in der Spalte X oder in der Spalte AD einer der vier Tabellen Zahlen ungleich Null auf. Genau diese Tabelle wollen wir in den Bereich **AF5:AI7** unseres Tabellenblattes ‚Direct comparisons‘ kopieren.

Dazu müssen wir wissen, welches die richtigen Zeilennummern sind.

In die Zelle **AA5** tragen wir folgende Formel ein:

=WENN(SVERWEIS(\$C5;GrpA!\$P\$18:\$X\$21;9;0)+SVERWEIS(\$C5;GrpA!\$P\$18:\$AD\$21;15;0)>0; VERGLEICH(\$C5;GrpA!\$P\$18:\$P\$21;0);0)

Der Ausdruck **SVERWEIS(\$C5;GrpA!\$P\$18:\$X\$21;9;0)** schaut für die betreffende Mannschaft (**\$C5**) auf dem Tabellenblatt "GrpA" in der Spalte X nach, ob dort eine Null steht (kein direkter Vergleich) oder eine Zahl größer als Null (direkter Vergleich).

Der Ausdruck **SVERWEIS(\$C5;GrpA!\$P\$18:\$AD\$21;15;0)** schaut entsprechend in der Spalte AD nach.

Der Ausdruck **VERGLEICH(\$C5;GrpA!\$P\$18:\$P\$21;0)** sagt uns, an welcher Position sich die betreffende Mannschaft (**\$C5**) in dem Bereich **GrpA!\$P\$18:\$P\$21** befindet.

Die Formel in **AA5** gibt also die Position der betreffenden Mannschaft innerhalb der ersten der vier möglichen Dreiergruppen zurück, wenn ein direkter Vergleich vorliegt, ansonsten eine Null.

Auf dieselbe Weise geben die Formeln in den Zellen **AB5**, **AC5** und **AD5** die Position der betreffenden Mannschaft in den anderen drei möglichen Dreiergruppen zurück. Diese Formeln lauten:

Zelle **AB5**:

**=WENN(SVERWEIS(\$C5;GrpA!\$P\$23:\$X\$26;9;0)+SVERWEIS(\$C5;GrpA!\$P\$23:\$AD\$26;15;0)>0;
VERGLEICH(\$C5;GrpA!\$P\$23:\$P\$26;0);0)**

Zelle **AC5**:

**=WENN(SVERWEIS(\$C5;GrpA!\$P\$28:\$X\$31;9;0)+SVERWEIS(\$C5;GrpA!\$P\$28:\$AD\$31;15;0)>0;
VERGLEICH(\$C5;GrpA!\$P\$28:\$P\$31;0);0)**

Zelle **AD5**:

**=WENN(SVERWEIS(\$C5;GrpA!\$P\$33:\$X\$36;9;0)+SVERWEIS(\$C5;GrpA!\$P\$33:\$AD\$36;15;0)>0;
VERGLEICH(\$C5;GrpA!\$P\$33:\$P\$36;0);0)**

Die Formeln in den Zelle **AA5**, **AB5**, **AC5** und **AD5** kopieren wir nach unten bis zur Zeile 8.

In der Zeile 9 bilden wir die Summe der jeweiligen Spalte:

Zelle **AA9**: **=SUMME(AA5:AA8)**

Zelle **AB9**: **=SUMME(AB5:AB8)**

Zelle **AC9**: **=SUMME(AC5:AC8)**

Zelle **AD9**: **=SUMME(AD5:AD8)**

Diese vier Zahlen sind dann entweder alle Null oder eine von ihnen ist eine Sechs.

In die Zelle **AE9** tragen wir folgende Formel ein:

{=MAX(WENN(\$AA9:\$AD9>0;SPALTE(\$AA9:\$AD9)-SPALTE(Z\$1);0))}

Dies ist eine sog. Matrixformel. Die geschweiften Klammern darf man nicht mit eingeben. Man muss aber die Eingabe der Formel mit **Strg-Umschalt-Return** abschließen. Dann werden die geschweiften Klammern automatisch gesetzt.

Diese Matrixformel berechnet die vier Ausdrücke

WENN(\$AA9>0;SPALTE(\$AA9)-SPALTE(Z\$1);0)

WENN(\$AB9>0;SPALTE(\$AB9)-SPALTE(Z\$1);0)

WENN(\$AC9>0;SPALTE(\$AC9)-SPALTE(Z\$1);0)

WENN(\$AD9>0;SPALTE(\$AD9)-SPALTE(Z\$1);0)

und gibt davon das Maximum zurück.

In der folgenden Abbildung sehen wir zum Beispiel in der Zelle **AB9** eine Sechs.

	Y	Z	AA	AB	AC	AD	AE
1							
2							
3	Clear?		Row numbers				Col num
4							
5	0		0	0	0	0	
6	0		0	1	0	0	
7	0		0	2	0	0	
8	0		0	3	0	0	
9			0	6	0	0	2
10							

Der erste, dritte und der vierte Ausdruck ergeben 0. Nur beim zweiten Ausdruck ist die Bedingung **\$AB9>0** wahr, so dass hier der Ausdruck **SPALTE(\$AB9)-SPALTE(Z\$1)** berechnet wird. Dieser Ausdruck ergibt 28 - 26 = 2. Das Maximum der vier Zahlen 0, 0, 2, 0 ist dann gleich 2.

Die Zahl in Zelle AE9 sagt uns, dass hier bei diesem Beispiel in der Spalte 2 der kleinen Tabelle Zeilennummern größer als 0 stehen.

Damit wissen wir, dass wir die Positionen bzw. die relativen Zeilennummern der Mannschaften in Spalte 2 der kleinen Tabelle **AA5:AD8** finden. Das werden wir in den Formeln des Bereichs **P5:T8** verwenden.

Außerdem wissen wir dadurch, dass wir vom Tabellenblatt „GrpA“ von den vier Tabellen im Bereich **Q18:T36** die zweite Tabelle auf das Tabellenblatt „Direct comparisons“ kopieren müssen. Und dies werden wir als nächstes tun. Wir verwenden hier wieder eine Matrix-Formel, diesmal nicht in einer einzelnen Zelle, sondern in einem Bereich.

Dazu markieren wir auf dem Tabellenblatt „Direct comparisons“ den Bereich **AF5:A17** und geben in der Bearbeitungszeile oben die folgende Formel ein:

```
=WENN($AA9>0;GrpA!$Q$18:$T$20;WENN($AB9>0;GrpA!$Q$23:$T$25;WENN($AC9>0;GrpA!$Q$28:$T$30;WENN($AD9>0;GrpA!$Q$33:$T$35;"0"))))
```

Die Eingabe muss unbedingt mit **Strg-Umschalt-Return** abgeschlossen werden. Die Formel wird dann automatisch in geschweifte Klammern eingeschlossen.

Die Formel bewirkt, dass bei unserem Beispiel die **zweite** Tabelle auf das Tabellenblatt „Direct comparisons“ kopiert wird.

Die Formel besteht aus ineinander geschachtelten WENN-Funktionen. Wenn **\$AA9>0** ist, soll der Bereich **GrpA!\$Q\$18:\$T\$20** kopiert werden, wenn **\$AB9>0** ist, der Bereich **GrpA!\$Q\$23:\$T\$25** usw., ansonsten wird überall "0" eingetragen.

Mit Hilfe der Matrix-Formel brauchen wir für den gesamten Bereich (die sog. Matrix) nur eine einzige Formel einzugeben. Die Alternative wäre, für jede der 12 Zellen eine Formel der Art

```
=WENN($AA9>0;GrpA!$Q$18;WENN($AB9>0;GrpA!$Q$23;WENN($AC9>0;GrpA!$Q$28;WENN($AD9>0;GrpA!$Q$33;"0"))))
```

einzugeben.

Auf dem **Tabellenblatt ‚Direct comparisons‘** wollen wir in den Zellen **W5** bis **W8** für jede Mannschaft mit ungeklärtem Rang einen roten Punkt eintragen. Als Vorbereitung dafür erzeugen wir in den Zellen **Y5** bis **Y8** erst einmal Nullen bzw. Einsen. In die Zellen **Y5** bis **Y8** tragen wir folgende Formeln ein:

Zelle **Y5**:

```
=WENN(UND(ODER(KÜRZEN(GrpA!$V$11;4)=KÜRZEN(GrpA!$V$12;4);KÜRZEN(GrpA!$V$11;4)=KÜRZEN(GrpA!$V$13;4);KÜRZEN(GrpA!$V$11;4)=KÜRZEN(GrpA!$V$14;4));SVERWEIS(GrpA!$P$11;GrpA!$P$4:$AC$7;14;0));1;0)
```

Zelle **Y6**:

```
=WENN(UND(ODER(KÜRZEN(GrpA!$V$12;4)=KÜRZEN(GrpA!$V$11;4);KÜRZEN(GrpA!$V$12;4)=KÜRZEN(GrpA!$V$13;4);KÜRZEN(GrpA!$V$12;4)=KÜRZEN(GrpA!$V$14;4));SVERWEIS(GrpA!$P$12;GrpA!$P$4:$AC$7;14;0));1;0)
```

Zelle **Y7**:

```
=WENN(UND(ODER(KÜRZEN(GrpA!$V$13;4)=KÜRZEN(GrpA!$V$11;4);KÜRZEN(GrpA!$V$13;4)=KÜRZEN(GrpA!$V$12;4);KÜRZEN(GrpA!$V$13;4)=KÜRZEN(GrpA!$V$14;4));SVERWEIS(GrpA!$P$13;GrpA!$P$4:$AC$7;14;0));1;0)
```

Zelle **Y8**:

```
=WENN(UND(ODER(KÜRZEN(GrpA!$V$14;4)=KÜRZEN(GrpA!$V$11;4);KÜRZEN(GrpA!$V$14;4)=KÜRZEN(GrpA!$V$12;4);KÜRZEN(GrpA!$V$14;4)=KÜRZEN(GrpA!$V$13;4));SVERWEIS(GrpA!$P$14;GrpA!$P$4:$AC$7;14;0));1;0)
```

Wenn die auf vier Nachkommastellen gekürzte Rangzahl der betreffenden Mannschaft mit der gekürzten Rangzahl einer anderen Mannschaft übereinstimmt, wird eine 1 eingetragen, sonst eine 0.

Die UND-Funktion **UND(ODER(...);SVERWEIS(GrpA!\$P\$11;GrpA!\$P\$4:\$AC\$7;14;0))** bewirkt, dass nur dann eine 1 eingetragen wird, wenn für die betreffende Mannschaft mindestens ein Spiel eingetragen ist. Der Ausdruck **SVERWEIS(GrpA!\$P\$11;GrpA!\$P\$4:\$AC\$7;14;0)** liest aus der Spalte AC des Tabellenblattes "GrpA" einen booleschen Wert, der angibt, ob die Anzahl der eingetragenen Spiele für die betreffende Mannschaft größer als Null ist.

Auch bei diesen Formeln ist es wichtig, dass vor den Zahlen 11, 12, 13 und 14 das Dollarzeichen steht!

10.3 Sichtbarer Bereich - Teil 2

In die Zelle **W5** tragen wir die Formel **=WENN(\$Y5>0;"•";"")** ein und kopieren sie nach unten bis zur Zelle **W8**.

Das Sonderzeichen kann man z. B. in eine beliebige Zelle mit Einfügen → Symbol eintragen und dann mit Kopieren + Einfügen in die Bearbeitungszeile einfügen.

Für die Mannschaften mit ungeklärtem Rang erscheint ein roter Punkt.

Im Bereich **P5:T8** soll die Tabelle für den **direkten Vergleich dreier Mannschaften** zu sehen sein.

In die Zelle **P5** tragen wir die folgende Formel ein:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;1;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;1;$AE9);1)))
```

Erklärung der Formel:

WENN(\$AE9=0;"") bedeutet: wenn kein direkter Vergleich dreier Mannschaften vorliegt, wird nichts eingetragen.

Die INDEX-Funktion in der Form **INDEX (Bereich; Zeile; Spalte)** gibt den Inhalt einer Zelle aus dem angegebenen Bereich zurück. In **Zeile** und **Spalte** stehen dabei die relativen Positionen der Zelle in diesem Bereich. So würde beispielsweise der Ausdruck **INDEX(C4:H9;2;3)** den Inhalt der Zelle **E5** zurückgeben.

Wenn ein direkter Vergleich vorliegt, steht in der Zelle **AE9** eine Zahl größer als Null und in der kleinen Tabelle **AA5:AD8** enthält eine der vier Spalten die Zeilennummern der drei Mannschaften, die an dem direkten Vergleich beteiligt sind. In der Zelle **AE9** steht die Nummer der Spalte, in der wir diese Zeilennummern finden können.

Der Ausdruck **WENN(INDEX(\$AA5:\$AD8;1;\$AE9)=0** prüft also, ob die Zeilennummer in der ersten Zeile gleich

Null ist. Wenn dies zutrifft, handelt es sich um die Mannschaft, die nicht am direkten Vergleich teilnimmt, und es wird nichts eingetragen. Im anderen Fall wird der Ausdruck **INDEX(\$AF5:\$AI7;INDEX(\$AA5:\$AD8;1;\$AE9);1)** ausgewertet und das Ergebnis eingetragen.

In diesem Ausdruck haben wir zwei verschachtelte INDEX-Funktionen.

Die innere INDEX-Funktion **INDEX(\$AA5:\$AD8;1;\$AE9)** liefert uns die relative Zeilennummer der betreffenden Mannschaft, die äußere INDEX-Funktion liefert uns aus Spalte 1 der Tabelle **AF5:A17** die **Punktzahl** der betreffenden Mannschaft.

Setzen wir für die Spaltennummer statt einer „1“ eine „2“ ein, erhalten wir die **Tordifferenz** und bei einer „3“ bzw. einer „4“ die **Tore**. Dies nutzen wir bei den Formeln in den Zellen **Q5**, **R5** und **T5**. Sie lauten:

Zelle **Q5**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;1;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;1;$AE9);2)))
```

Zelle **R5**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;1;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;1;$AE9);3)))
```

Zelle **T5**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;1;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;1;$AE9);4)))
```

Die restlichen Formeln in der Tabelle **P5:T8** sehen alle ganz ähnlich aus:

Zelle **P6**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;2;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;2;$AE9);1)))
```

Zelle **Q6**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;2;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;2;$AE9);2)))
```

Zelle **R6**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;2;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;2;$AE9);3)))
```

Zelle **T6**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;2;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;2;$AE9);4)))
```

Zelle **P7**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;3;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;3;$AE9);1)))
```

Zelle **Q7**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;3;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;3;$AE9);2)))
```

Zelle **R7**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;3;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;3;$AE9);3)))
```

Zelle **T7**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;3;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;3;$AE9);4)))
```

Zelle **P8**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;4;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;4;$AE9);1)))
```

Zelle **Q8**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;4;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;4;$AE9);2)))
```

Zelle **R8**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;4;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;4;$AE9);3)))
```

Zelle **T8**:

```
=WENN($AE9=0;"";WENN(INDEX($AA5:$AD8;4;$AE9)=0;"";INDEX($AF5:$AI7;INDEX($AA5:$AD8;4;$AE9);4)))
```


Damit die Gruppen, in denen ein direkter Vergleich durchgeführt wird, noch besser ins Auge fallen, soll links vom Gruppennamen ein großer blauer Punkt erscheinen. In die Zelle **B3** tragen wir die folgende Formel ein:
=WENN(ODER(\$AE9>0;\$L5<>"";\$L6<>"";\$L7<>"");"●";"")

Das Sonderzeichen kann man wieder in eine beliebige Zelle mit **Einfügen** → **Symbol** eintragen und dann mit **Kopieren + Einfügen** in die Bearbeitungszeile einfügen. Es wird evtl. in der Bearbeitungszeile als Leerzeichen angezeigt, funktioniert aber trotzdem.

Wir markieren nun den Bereich **B3:A19** und kopieren ihn z. B. mit **Strg C**.

Dann klicken wir auf die Zelle **B11** und fügen die Kopie ein z. B. mit **Strg V**.

Wichtig: Der neue Bereich ist noch markiert!

Mit **Suchen und Ersetzen** ändern wir in allen Formeln des markierten Bereichs den Text „GrpA“ in „GrpB“.

In Zelle **C11** ändern wir die Formel noch zu **=Language!\$E\$93&" B"**.

Die Bereiche für die restlichen Gruppen erzeugen wir auf die gleiche Weise. - **Alles geschafft!**

Der fertige Spielplan ist in der Datei „**EURO_Tut_final.xlsx**“ zu finden.

11 Anpassung des Spielplans für eine neue Europameisterschaft

Der Spielplan ist so angelegt, dass er möglichst einfach und benutzerfreundlich an das nächste Turnier angepasst werden kann.

Wer ganz genau sein möchte, aktualisiert die auf dem Tabellenblatt ‚Language‘ in der Spalte D die Rangzahlen entsprechend der neuen Rangreihenfolge am Ende der Qualifikationsphase. Dabei müssen keine Zeilen vertauscht werden! Die Rangzahlen können auch ungeordnet sein – Hauptsache, sie sind eindeutig. Diesen Aufwand kann man sich jedoch auch sparen, denn man kann eine unklare Reihenfolge in der Gruppenphase immer durch die Eingabe eines Fair-Play-Bonus klären.

Folgende Schritte sind im Einzelnen durchzuführen:

11.1 Tabellenblatt 'Language'

Hier müssen nur die Austragungsorte aktualisiert werden.

11.2 Lokalzeit des Gastgeberlandes

Auf dem Tabellenblatt ‚TimeZone‘ wird in der Zelle L5 die Zeitzone des Gastgeberlandes aktualisiert. Dazu müssen der Blattschutz aufgehoben und die ausgeblendeten Spalten eingeblendet werden.

11.3 Zuordnung der Teams zu den Gruppen

Das Tabellenblatt ‚Groups‘ wird eingeblendet. In der Spalte C werden die Nummern der Teams eingetragen, die der betreffenden Gruppe zugeordnet wurden. Für Team A1 wird zum Beispiel die Nummer 14 eingetragen, da die Türkei auf dem Tabellenblatt ‚Language‘ die Nummer 14 besitzt usw.

11.4 Aktualisierung der Match-Daten

Das Tabellenblatt ‚Matches‘ wird eingeblendet. Hier müssen in der Spalte E die Anstoßzeiten und in der Spalte G die Nummern der Austragungsorte aktualisiert werden.

Sollte eine Anpassung der Match-Nummern an den UEFA-Spielplan notwendig sein, können nach Aufhebung des Blattschutzes in der Spalte B die entsprechenden Änderungen vorgenommen werden.

11.5 Aktualisierung der Matchnummern auf dem Spielplan

Auf dem Tabellenblatt ‚EURO‘ werden nun die Match-Nummern angepasst. Es empfiehlt sich, die sechs Gruppenspiele innerhalb jeder Gruppe in der zeitlichen Reihenfolge anzuordnen.

Das war's! - Alle diese Änderungen werden auf den restlichen Tabellenblättern automatisch übernommen.

Anhang: Spezialfall ‚Direkte Vergleiche‘

Nach den UEFA-Regeln für die Qualifikation zum Achtelfinale haben direkte Vergleiche Vorrang vor der Tordifferenz und der Anzahl erzielter Tore. Wenn nach einem direkter Vergleich immer noch Mannschaften mit Gleichstand übrig bleiben, wird der direkte Vergleich erneut auf diese Mannschaften angewendet.

Dieser Spezialfall zeigt ein Beispiel, in dem nach dem direkten Vergleich aller vier Mannschaften ein weiterer direkter Vergleich dreier Mannschaften und danach ein dritter direkter Vergleich zweier Mannschaften stattfindet.

Die Spielergebnisse sehen folgendermaßen aus:

Niederlande	Ukraine	2 : 2
Österreich	Nordmazedonien	1 : 1
Niederlande	Österreich	1 : 1
Ukraine	Nordmazedonien	0 : 0
Nordmazedonien	Niederlande	1 : 1
Ukraine	Österreich	0 : 0

Es ergibt sich folgende Gruppentabelle:

	Pkte.	TD	Tore
Niederlande	3	0	4 : 4
Österreich	3	0	2 : 2
Nordmazedonien	3	0	2 : 2
Ukraine	3	0	2 : 2

Der direkte Vergleich aller vier Mannschaften ergibt: die Niederlande stehen auf Platz 1.

Für die drei anderen Mannschaften wird ein weiterer direkter Vergleich durchgeführt, der nur die Spiele dieser drei gegeneinander berücksichtigt. Für diesen direkten Vergleich ergibt sich folgende Tabelle:

	Pkte.	TD	Tore
Österreich	2	0	1 : 1
Nordmazedonien	2	0	1 : 1
Ukraine	2	0	0 : 0

Damit steht die Ukraine an letzter Stelle.

Für Österreich und Nordmazedonien findet ein erneuter direkter Vergleich statt.

Austria	North Macedonia	1 : 1
---------	-----------------	-------

Dieser führt aber zu keiner Entscheidung, da die Direktbegegnung zwischen diesen beiden unentschieden endete. Da die Anzahl der Siege aus allen Gruppenspielen auch gleich ist, entscheidet als nächstes die Fair-Play-Wertung. Bei gleicher Fair-Play-Wertung entscheidet dann endgültig der Rang in der Qualifikationsrunde. Da steht Österreich auf Rang 16 vor Nordmazedonien auf Rang 30 (EURO 2020/2021).

